



Manual Excel 2010

Cálculos y Funciones

CONTENIDO**ESCRIBIR FÓRMULAS UTILIZANDO OPERADORES**

PRECEDENCIA DE LOS OPERADORES MATEMÁTICOS

REFERENCIAR CELDAS O RANGOS DE CELDAS**COPIAR Y PEGAR FÓRMULA**

REFERENCIA RELATIVA

REFERENCIA ABSOLUTA

REFERENCIA MIXTA

REFERENCIA POR NOMBRE

APLICAR CAMBIOS DE NOMBRE A REFERENCIAS

CÁLCULOS

CÁLCULO DE PORCENTAJES

CÁLCULO DE TOTALES ACUMULADOS

CÁLCULOS CON FECHAS

CÁLCULO ENTRE CELDAS O RANGOS DISPUESTOS EN OTRA HOJA

USO DE OPERACIONES DE COMPARACIÓN**USO DE OPERADORES DE TEXTO: &****FUNCIONES EN EXCEL**

SINTAXIS BÁSICA

FUNCIÓN SUMA

FUNCIÓN PROMEDIO

FUNCIÓN MÁXIMO

FUNCIÓN MÍNIMO

FUNCIÓN CONTARA

FUNCIÓN CONTAR

FUNCIÓN CONTAR.BLANCO

FUNCIONES LÓGICAS

FUNCIÓN SI

FUNCIÓN Y

FUNCIÓN O

FUNCIÓN CONTAR.SI

FUNCIÓN SUMAR.SI

FUNCIÓN PI (OPCIONAL)

FUNCIÓN PRODUCTO (OPCIONAL)

FUNCIÓN POTENCIA (OPCIONAL)

FUNCIÓN NO (OPCIONAL)

FUNCIONES DE FECHA Y HORA

FUNCIÓN HOY

FUNCIÓN AÑO

FUNCIÓN MES

FUNCIÓN SIFECHA
FUNCIÓN AHORA
FUNCIÓN DIASEM
FUNCIÓN ENTERO
FUNCIÓN REDONDEAR
FUNCIÓN TRUNCAR

FUNCIONES DE BÚSQUEDA Y REFERENCIA:

FUNCIÓN BUSCARV
FUNCIÓN BUSCARH
FUNCIÓN ELEGIR
FUNCIÓN CONCATENAR
FUNCIÓN NOMPROPIO (OPCIONAL)
FUNCIÓN DERECHA
FUNCIÓN IZQUIERDA
FUNCIÓN ESPACIOS
FUNCIÓN EXTRAE
FUNCIÓN LARGO
FUNCIÓN HALLAR

FUNCIONES DE INFORMACIÓN

FUNCIÓN ESERROR
FUNCIÓN ESBLANCO
FUNCIÓN ESNUMERO
FUNCIÓN JERARQUÍA
FUNCIÓN K.ESIMO.MAYOR
FUNCIÓN K.ESIMO.MENOR
FUNCIÓN SUMAR.SI.CONJUNTO
FUNCIÓN CONTAR.SI.CONJUNTO

ANIDAMIENTO DE FUNCIONES

Uso de operadores matemáticos: +, -, *, /

La planilla de cálculo es una herramienta muy útil que nos permite realizar cálculos y operaciones con distintos tipos de datos.

Los cálculos que podemos realizar en una hoja de cálculo los llevamos a cabo mediante fórmulas. Si en la fórmula se introducen referencias a celdas en lugar de valores en número, podremos ver actualizados los cálculos automáticamente cuando modifiquemos un dato en esa referencia.

En la siguiente planilla, podemos observar algunos cálculos que utilizamos frecuentemente, ellos son:

Suma (+)

Para realizar dicho cálculo escribimos, por ejemplo, en C9:

$$= C3+C4+C5+C6+C7+C8$$

Producto (*)

Para realizar dicho cálculo escribimos, por ejemplo, en D3:

$$= B3 * C3$$

Resta (-)

Para realizar dicho cálculo escribimos, por ejemplo, en F3

$$= D3 - E3$$

División: (/), Potenciación (^)

(No están presentes en esta planilla)

Porcentaje de un valor dado (%)

Para realizar dicho cálculo escribimos, por ejemplo, en E3

$$= D3 * 5 \%.$$

E3						
	A	B	C	D	E	F
	COMPUMAR					
1						
2	Producto	Cantidad	Precio Unitario	Precio	Descuento 5%	Precio Final
3	Memoria 1 GB Kingston	10	\$ 119,00	\$ 1.190,00	\$ 59,50	\$ 1.130,50
4	Disco rígido Samsung 320 GB	5	\$ 299,00	\$ 1.495,00		
5	Pen Drive Novatech 4GB	8	\$ 109,00	\$ 872,00		
6	Estabilizador Concept 500VA	12	\$ 109,00	\$ 1.308,00		
7	Disco rígido externo IOMEGA	4	\$ 739,00	\$ 2.956,00		
8	Mouse Wireless Notebook 4000	8	\$ 229,00	\$ 1.832,00		
9	TOTALES		\$ 1.604,00			

Resumiendo, para ejecutar las operaciones matemáticas básicas como suma, resta o multiplicación, combinar números y generar resultados numéricos, utilizamos los siguientes operadores matemáticos.

Operador	Operación
+	Suma
-	Resta
*	Multiplicación
/	División
%	Porcentaje
^	Potenciación

Escribir fórmulas utilizando operadores

Podemos observar que las fórmulas en Excel siempre comienzan con un signo igual (=). El signo igual indica a Excel que los caracteres siguientes constituyen una fórmula. Detrás del signo igual están los elementos que se van a calcular (los operandos), separados por operadores de cálculo. Excel calcula la fórmula de izquierda a derecha, según el orden específico de cada operador de la fórmula.

Precedencia de los operadores matemáticos

Cuando combinamos varios operadores en una única fórmula, las operaciones se ejecutarán en el orden que indicamos a continuación:

Orden	Operador	Operación
1	%	Porcentaje
2	^	Potenciación
3	*	Multiplicación
3	/	División
4	+	Suma
4	-	Resta

Si una fórmula contiene operadores con la misma precedencia (por ejemplo, si una fórmula contiene un operador de multiplicación y otro de división), Excel evaluará los operadores de izquierda a derecha.

Para cambiar el orden de evaluación, escribimos entre paréntesis la parte de la fórmula que se calculará en primer lugar. Por ejemplo, la siguiente fórmula da como resultado 11 porque Excel calcula la multiplicación antes que la suma. La fórmula multiplica 2 por 3 y, a continuación, suma 5 al resultado.

=5+2*3

Por el contrario, si utilizamos paréntesis para cambiar la sintaxis, Excel sumará 5 + 2 y, a continuación, multiplica el resultado por 3, con lo que se obtiene 21.

=(5+2)*3

Referenciar celdas o rangos de celdas

Para referenciar **una celda** dispuesta en la misma hoja escribimos el nombre de la columna a la que pertenece y a continuación el número de fila, por ejemplo, la primera celda pertenece a la columna A y a la fila 1 por lo tanto para referenciar dicha celda escribimos **=A1**.

Para referenciar **un rango continuo**, dispuesto en la misma hoja, escribimos el nombre de la primera celda que lo compone, dos puntos (:), y a continuación el nombre de la última celda, por ejemplo, **A2:B3**. Dicho rango está compuesto por las celdas A2, A3, B2, B3.

Para referenciar **un rango de celdas separadas**, dispuesto en la misma hoja, escribimos el nombre de cada celda separadas con una coma (,).

Para referenciar **dos o más rangos discontinuos** dispuestos en la misma hoja, escribimos el nombre de cada rango separados por punto y coma (;).

Para referenciar celdas o rango de celdas dispuestos en otra hoja debemos escribir primero el nombre de la hoja en la cual está incluido dicho rango, seguido del signo !.

Por ejemplo:

Para referenciar, en la Hoja1, la celda B3 incluida en la Hoja2, escribimos:

Hoja2!B3.

Para referenciar, en la Hoja1, el rango A1:B3 incluido en la Hoja2, escribimos: **Hoja2!A1:B3**.

Copiar y pegar fórmula

Para completar la planilla que figura en la página 2 debemos copiar las fórmulas ingresadas en la primera fila al resto del rango.

	D	E
Precio	\$ 1.190,00	Descuento 5%

Para copiar una fórmula a otras celdas se utiliza,

generalmente, la **técnica de arrastrar**. Para utilizar dicha técnica debemos ingresar la fórmula en la primera celda del rango a calcular, pulsamos **ENTER** para obtener el resultado. Luego hacemos clic en ella, posicionamos el mouse donde muestra la imagen (ángulo inferior derecho de dicha celda), y cuando éste toma la forma de cruz fina, hacemos clic y arrastramos en la dirección de las celdas a copiar.

Otro modo de copiar fórmulas a otras celdas es el siguiente:

Una vez obtenido el resultado en la primera celda, la seleccionamos y pulsamos **Copiar**. Luego pulsamos en la celda a partir de la cual deseamos pegar la fórmula y pulsamos **Pegar**.

Un tercer modo de copiar fórmulas a otras celdas es utilizando la pestaña **Inicio**, grupo **Modificar**, botón **Rellenar**:

Una vez obtenido el resultado en la primera celda, seleccionamos dicha celda junto al rango donde deseamos copiar la fórmula. Luego recurrimos a la pestaña **Inicio**, grupo **Modificar**, botón **Rellenar** y escogemos **Hacia abajo** (si queremos copiar la fórmula en la misma columna) o **Hacia la derecha** (si queremos copiar la fórmula en la misma fila).



Volviendo al primer ejemplo, para copiar la fórmula que figura en C9 a las celdas D9, E9 y F9, como pertenecen a la misma fila que C9 escogemos en el menú anterior **Hacia la derecha**.

REFERENCIA RELATIVA

Como mencionamos, para copiar una fórmula a otras celdas utilizamos, por ejemplo, la **técnica de arrastrar**.

Formula bar: $=B3*C3$

	A	B	C	D	E	F
	COMPUMAR					
1						
2	Producto	Cantidad	Precio Unitario	Precio	Descuento 5%	Precio Final
3	Memoria 1 GB Kingston	10	\$ 119,00	\$ 1.190,00	\$ 59,50	\$ 1.130,50
4	Disco rígido Samsung 320 GB	5	\$ 299,00			
5	Pen Drive Novatech 4GB	8	\$ 109,00			
6	Estabilizador Concept 500VA	12	\$ 109,00			
7	Disco rígido externo IOMEGA	4	\$ 739,00			
8	Mouse Wireless Notebook 4000	8	\$ 229,00			
9	TOTALES		\$ 1.604,00			

En el ejemplo, en **D3** ingresamos $= B3 * C3$ y luego copiamos la misma al resto del rango. Al copiar una fórmula es indispensable considerar la dirección de la celda que ocupa.

Si posicionamos el cursor en **D4**, por ejemplo, comprobaremos que la fórmula ingresada se transforma en $= B4 * C4$, es decir, **B3** se actualiza a **B4** y **C3** se actualiza a **C4**, y así sucesivamente con el resto de las celdas.

Dichas celdas reciben el nombre de **REFERENCIAS RELATIVAS**.

Formula bar: $=B4*C4$

	B	C	D
	COMPUMAR		
	Cantidad	Precio Unitario	Precio
on	10	\$ 119,00	\$ 1.190,00
320 GB	5	\$ 299,00	\$ 1.495,00
GB	8	\$ 109,00	\$ 872,00
: 500VA	12	\$ 109,00	\$ 1.308,00
MEGA			

Pero en determinadas situaciones, es necesario hacer referencia a una determinada celda y no deseamos que su dirección varíe al copiar la fórmula.

En la siguiente planilla se hizo un registro, sobre un total de 468 personas, de la cantidad de personas que son simpatizantes de distintos clubes de fútbol (Columna B). En la celda B10 figura el total de personas encuestadas.

Formula bar: $=B2/B10$

	A	B	C
	EQUIPO	SIMPATIZANTES	PORCENTAJES
1			
2	BOCA	89	19,02%
3	RIVER	45	
4	INDEPENDIENTE	56	
5	RACING	32	
6	HURACAN	107	
7	ARGENTINOS	34	
8	FERRO	56	
9	VELEZ	49	
10		468	
11			

En la columna C se calculó qué porcentaje representa cada valor del total. Si razonamos por medio de una regla de tres simples resulta:

468 ----- 100%

89 ----- x %, El valor de **x** se obtiene de $= \frac{89}{468} * 100 \%$

468

Como en Excel tenemos la facilidad de otorgar a un rango de celdas el formato más apropiado, en este caso le asignamos formato porcentaje, con lo cual no es necesario multiplicar por 100 y agregarle el signo %.

En resumen la fórmula a ingresar es **= B2 / B10**.

C2			f _x	=B2/B10
	A	B	C	
	EQUIPO	SIMPATIZANTES	PORCENTAJES	
1				
2	BOCA	89	19,02%	
3	RIVER	45	#DIV/0!	
4	INDEPENDIENTE	56	#DIV/0!	
5	RACING	32	#DIV/0!	
6	HURACAN	107	#DIV/0!	
7	ARGENTINOS	34	#DIV/0!	
8	FERRO	56	#DIV/0!	
9	VELEZ	49	#DIV/0!	
10		468		

Podemos comprobar que, al arrastrar la fórmula ingresada en la primera celda al resto del rango no obtenemos el resultado esperado.

Para entender que ha sucedido posicionemos el cursor en alguna de las celdas donde obtuvimos un resultado erróneo, y miremos en la barra de fórmula qué celdas figuran en la fórmula.

La fórmula que figura en la barra de fórmulas al seleccionar, por ejemplo,

la celda **C3** es **=B3 / B11**, y **B11 se encuentra vacía!**, es decir intentamos dividir por 0. Por tal motivo, Excel nos muestra como resultado de la operación **#iDIV/0!**

C3			f _x	=B3/B11
	A	B	C	
	EQUIPO	SIMPATIZANTES	PORCENTAJES	
1				
2	BOCA	89	19,02%	
3	RIVER	45	#DIV/0!	
4	INDEPENDIENTE	56	#DIV/0!	
5	RACING	32	#DIV/0!	
6	HURACAN	107	#DIV/0!	
7	ARGENTINOS	34	#DIV/0!	
8	FERRO	56	#DIV/0!	
9	VELEZ	49	#DIV/0!	
10		468		

REFERENCIA ABSOLUTA

Para resolver dicho inconveniente es necesario indicarle a Excel que considere siempre, para realizar dicho cálculo, la celda **B10**. Dicho, en otros términos, que deje "fija" dicha celda.

Para ello debemos agregar delante del identificador de fila y delante del identificador de columna el signo \$.

Finalmente la fórmula que ingresamos es **=B2/\$B\$10**

La celda **\$B\$10** recibe el nombre de **REFERENCIA ABSOLUTA**.

La referencia **ABSOLUTA** especifica la dirección exacta de una celda, independientemente de la posición de la celda que contiene la fórmula.

	C2		f _x	=B2/\$B\$10
	A	B	C	
1	EQUIPO	SIMPATIZANTES	PORCENTAJES	
2	BOCA	89	19.02%	
3	RIVER	45	9.62%	
4	INDEPENDIENTE	56	11.97%	
5	RACING	32	6.84%	
6	HURACAN	107	22.86%	
7	ARGENTINOS	34	7.26%	
8	FERRO	56	11.97%	
9	VELEZ	49	10.47%	
10		468		

REFERENCIA MIXTA

En la siguiente planilla generaremos una tabla de multiplicar, en la cual en cada celda mostraremos el resultado de multiplicar cada número que figura en la fila 2 por cada número que figura en la columna A.

Por ejemplo en **B3** ingresamos **= A3 * B2**.

B3														
	A	B	C	D	E	F	G	H	I	J	K			
	Tabla de multiplicar													
1														
2		1	2	3	4	5	6	7	8	9	10			
3	1	1												
4	2													
5	3													
6	4													
7	5													
8	6													
9	7													
10	8													
11	9													
12	10													

Observemos el resultado de arrastrar la fórmula ingresada verticalmente, evidentemente no es el resultado esperado, ya que, en la **columna B** debería mostrarse los números del 1 al 10, y en la **columna C** debería mostrarse los resultados 2, 4, 6, 8, 10....

Formula bar: $=A7*B6$

	A	B	C	D	E	F	G	H	I	J	K
1	Tabla de multiplicar										
2		1	2	3	4	5	6	7	8	9	10
3	1	1	2								
4	2	2	4								
5	3	6	24								
6	4	24	576								
7	5	120	69120								
8	6	720	5E+07								
9	7	5040	3E+11								
10	8	40320	1E+16								
11	9	362880	4E+21								
12	10	4E+06	1E+28								

Observemos el resultado de arrastrar la fórmula ingresada horizontalmente, evidentemente tampoco es el resultado esperado.

Formula bar: $=D4*E3$

	A	B	C	D	E	F	G	H	I	J	K
1	Tabla de multiplicar										
2		1	2	3	4	5	6	7	8	9	10
3	1	1	2	6	24	120	720	5040	40320	362880	4E+06
4	2	2	4	24	576	69120	5E+07	3E+11	1E+16	4E+21	1E+28
5	3										
6	4										
7	5										
8	6										
9	7										
10	8										
11	9										
12	10										

Analicemos qué celdas deberían multiplicarse en la celda seleccionada **E4**.

En **E4** debería mostrarse el siguiente cálculo: **=A4*E2**, sin embargo aparece **=D4*E3**.

Podemos concluir que debemos dejar fijo la columna A y fila 2, por tal razón ingresamos en B3 = **\$A3*B\$2** y luego arrastramos la fórmula en cualquiera de las dos direcciones (horizontal o verticalmente). El resultado será el esperado.

	A	B	C	D	E	F	G	H	I	J	K
1	Tabla de multiplicar										
2		1	2	3	4	5	6	7	8	9	10
3	1	1	2	3	4	5	6	7	8	9	10
4	2	2	4	6	8	10	12	14	16	18	20
5	3	3	6	9	12	15	18	21	24	27	30
6	4	4	8	12	16	20	24	28	32	36	40
7	5	5	10	15	20	25	30	35	40	45	50
8	6	6	12	18	24	30	36	42	48	54	60
9	7	7	14	21	28	35	42	49	56	63	70
10	8	8	16	24	32	40	48	56	64	72	80
11	9	9	18	27	36	45	54	63	72	81	90
12	10	10	20	30	40	50	60	70	80	90	100

Las celdas **\$A3**, **B\$2**,... reciben el nombre de **REFERENCIA MIXTA**, ya que en ellas sólo se fija la columna o la fila.

Para convertir una referencia relativa en absoluta o mixta, estando la celda seleccionada, pulsamos la **tecla de función F4**.

Pulsando una vez **F4** transforma la referencia en absoluta.

Pulsando por segunda vez la transforma en relativa, dejando fija la fila.

Pulsando por tercera vez la transforma en relativa, pero en esta ocasión, deja fija la columna.

	=\$B\$3	
	=B\$3	
	=\$B3	

REFERENCIA POR NOMBRE


Las personas estamos más acostumbradas a nombrar las cosas por su nombre que por números de referencia, lo vemos todo mucho más claro e intuitivo.

tivo. Podemos ver una fórmula relativamente sencilla pero si contiene muchas referencias nos puede costar interpretarla.

Por eso Excel nos facilita un poco las cosas permitiendo que podamos ponerles nombre a las celdas y a las fórmulas para así identificarlas más fácilmente.


Podemos dar **nombre** a una celda (o rango de celdas) de varias maneras.

1. Abriendo el **Administrador de Nombres** de la pestaña **Fórmulas**, grupo **Nombres definidos**.




Se abrirá el siguiente cuadro de diálogo:

Aquí encontraremos un listado de todos los nombres que hayamos creado y a qué celda o rango de celdas hacen referencia.



Para crear un nuevo nombre tenemos que hacer clic en el botón **Nuevo**.

Se abrirá el cuadro de diálogo Nombre nuevo como vemos en la imagen.



En el recuadro **Nombre**: escribimos el nombre que le queremos dar a la celda.

En **Hace referencia a**: escribimos la referencia de la celda (o rango) como vemos en la imagen (o directamente seleccionamos con el mouse).

Es importante escribir el signo igual y utilizar referencias absolutas (\$A\$1).




2. Escribiendo directamente el nombre que deseamos asignar, en el **Cuadro de nombres** y presionando ENTER.

Veamos un ejemplo práctico, una lista de costos de productos a los que hay que agregarle la ganancia deseada y el IVA para obtener el precio final de venta.

Ganancia lo asignamos directamente desde el **Cuadro de nombres**. **Costo** lo asignaremos desde **Asignar nombre a un rango** (ficha **Fórmulas**, grupo **Nombres definidos**).

Asignamos **Beneficio** al rango **C4:C7** usando el método que nos quede más cómodo. También asignamos **IVA** a **D4:D7**. En **D1** escribimos **IVA** y en **E1** ingresamos **21%**, a **E1** le damos nombre **Impuesto**. Veamos como trabaja este tipo de referencias:

Para calcular el beneficio por cada producto (se obtiene de multiplicar el costo del producto por el porcentaje de ganancia deseada, 45% en este caso), comenzamos por escribir, en **C4**, el signo **=** para introducir un cálculo. Escribimos ahora el **nombre** de la celda, para este ejemplo: **Costo**. Cuando comenzamos a escribir este nombre, aparece la lista con funciones y nombres disponibles para incluir en el cálculo, así que podemos darle doble clic directamente desde la lista.



Escribimos * para realizar la multiplicación y comenzamos a escribir el nombre de la celda que contiene el porcentaje de ganancia que deseamos aplicar, **Ganancia**. También aparece la lista que podemos aprovechar. Copiamos a las demás celdas y observamos que ya no dependemos de asignar referencia absoluta a **B1** ya que Excel identifica una celda con nombre independientemente de dónde se aplique ésta. Incluso se puede usar directamente en otras Hojas sin hacer referencia a ellas.

El Impuesto se multiplica a la suma de Costo y Beneficio. Finalmente, el precio de venta se obtiene de la suma de Costo, Beneficio e IVA.

Utilizando este tipo de referencias nos independizamos de las referencias relativas y absolutas.

CONTAR.SI.CONJUN... $\times \checkmark f_x$ =Costo*ga					
	A	B	C	D	E
1	Ganancia 45%			IVA 21%	
2					
3	Código Producto	Costo	Beneficio	IVA	Precio Venta
4	FG-478	\$ 32,00	=Costo*ga		
5	IL-369	\$ 15,80			
6	OP-014	\$ 47,00			
7	YU-003	\$ 21,66			
8					

C6 $\times \checkmark f_x$ =Costo*Ganancia					
	A	B	C	D	E
1	Ganancia 45%			IVA 21%	
2					
3	Código Producto	Costo	Beneficio	IVA	Precio Venta
4	FG-478	\$ 32,00	\$ 14,40		
5	IL-369	\$ 15,80	\$ 7,11		
6	OP-014	\$ 47,00	\$ 21,15		
7	YU-003	\$ 21,66	\$ 9,75		
8					

D5 $\times \checkmark f_x$ =(Costo+Beneficio)*Impuesto					
	A	B	C	D	E
1	Ganancia 45%			IVA 21%	
2					
3	Código Producto	Costo	Beneficio	IVA	Precio Venta
4	FG-478	\$ 32,00	\$ 14,40	\$ 9,74	
5	IL-369	\$ 15,80	\$ 7,11	\$ 4,81	
6	OP-014	\$ 47,00	\$ 21,15	\$ 14,31	
7	YU-003	\$ 21,66	\$ 9,75	\$ 6,60	
8					



E6 $\times \checkmark f_x$ =Costo+Beneficio+IVA					
	A	B	C	D	E
1	Ganancia 45%			IVA 21%	
2					
3	Código Producto	Costo	Beneficio	IVA	Precio Venta
4	FG-478	\$ 32,00	\$ 14,40	\$ 9,74	\$ 56,14
5	IL-369	\$ 15,80	\$ 7,11	\$ 4,81	\$ 27,72
6	OP-014	\$ 47,00	\$ 21,15	\$ 14,31	\$ 82,46
7	YU-003	\$ 21,66	\$ 9,75	\$ 6,60	\$ 38,00
8					

Aplicar cambios de nombre a referencias

Cuando asignamos nombres a diversas celdas, las fórmulas que hagan referencia a estas celdas no se cambian automáticamente sustituyendo la nomenclatura de columna-fila por el nombre.

Por ejemplo: Si tenemos la fórmula **=A1+B1** y posteriormente asociamos la celda A1 con el nombre Gasto1 y la celda B1 con el nombre Gasto2. En la fórmula continuará saliendo **=A1+B1**.

Para que los nombres se actualicen en todas las



referencias debemos ir a la pestaña Fórmulas, desplegar el menú que se encuentra en la opción Asignar un nombre a un rango y selecciona Aplicar nombres....

En el cuadro de diálogo Aplicar nombres que nos aparece debemos seleccionar los nombres que queramos aplicar y hacemos clic en Aceptar.

Cálculos

Cálculo de porcentajes

- ✎ Para calcular **un porcentaje de una cantidad dada**, multiplicamos la celda que contiene la cantidad por el porcentaje. Ejemplo: la fórmula ingresada en E3 ($= D3 * 5 \%$).
- ✎ Para calcular **qué porcentaje es una cantidad A de otra B**, primero al rango que guardará el resultado le aplicamos formato porcentaje, luego escribimos $= A / B$. Por último copiamos dicha fórmula al resto del rango.

Cálculo de totales acumulados

En la columna C de la siguiente planilla calcularemos el costo acumulado teniendo en cuenta los valores ingresados en la columna B.




Para ello en la primera celda **C2** debemos ingresar el valor del primer costo (B2), es decir en **C2** escribimos $=B2$

En **C3** debemos ingresar $=C2 + B3$

En **C4** debemos ingresar $= C3 + B4$ y así sucesivamente.

Es decir, en cada celda de la columna **C** debemos calcular la suma del

	A	B	C
1	DESCRIPCION	COSTO	COSTO ACUMULADO
2	Trabajos preliminares	\$ 6.492,00	\$ 6.492,00
3	Excavaciones	\$ 2.413,00	\$ 8.905,00
4	Mampostería	\$ 76.298,00	\$ 85.203,00
5	Hormigón armado	\$ 17.027,00	\$ 102.230,00
6	Aislaciones	\$ 6.207,00	\$ 108.437,00
7	Cubiertas	\$ 40.986,00	\$ 149.423,00
8	Revoques	\$ 38.402,00	\$ 187.825,00
9	Cielorrasos	\$ 9.324,00	\$ 197.149,00
10	Contrapisos	\$ 13.652,00	\$ 210.801,00
11	Pisos	\$ 15.223,00	\$ 226.024,00
12	Zócalos	\$ 1.841,00	\$ 227.865,00
13	Revestimiento	\$ 3.476,00	\$ 231.341,00
14	carpintería	\$ 23.193,00	\$ 254.534,00
15	Instalación sanitaria	\$ 22.248,00	\$ 276.782,00
16	Instalaciones gas	\$ 8.978,00	\$ 285.760,00
17	Instalaciones eléctricas	\$ 16.598,00	\$ 302.358,00
18	Pinturas	\$ 12.504,00	\$ 314.862,00
19	Cristales	\$ 4.596,00	\$ 319.458,00
20	Varios	\$ 8.890,00	\$ 328.348,00
21	Gastos de obra	\$ 36.226,00	\$ 364.574,00
22	Beneficio	\$ 36.460,00	\$ 401.034,00
23	TOTALES	\$ 401.034,00	

ELEGIR					=C3+B4
	A	B	C		
1	DESCRIPCION	COSTO	COSTO ACUMULADO		
2	Trabajos preliminares	\$ 6.492,00	\$ 6.492,00		
3	Excavaciones	\$ 2.413,00	\$ 8.905,00		
4	Mampostería	\$ 76.298,00	=C3+B4		
5	Hormigón armado	\$ 17.027,00			
6	Aislaciones	\$ 6.207,00			

contenido de la celda ubicada por encima de dicha celda y la celda ubicada a la izquierda de la misma.

Por tal motivo, una vez ingresada la fórmula en **C3** podemos utilizar la técnica de arrastrar para copiar la fórmula al resto del rango.

Además podemos verificar que el último costo acumulado (C22) coincide con el total de la columna Costo (B23).

Cálculos con fechas

Excel nos permite realizar cálculos con distintos tipos de datos, entre ellos, fechas.

La siguiente planilla corresponde al informe realizado en una empresa de transportes en la que desean calcular la fecha en la que deben realizar el cambio del filtro de aire y de aceite a su flota de camiones.

Para realizar dicho cálculo tienen en cuenta los tiempos especificados por el fabricante, la fecha en que se realizó el último cambio y la fecha en que se hizo el informe.

	A	B	C	D	E	F	G	H
1								
2		Filtro Aire	Filtro Aceite		Fecha Informe	10/05/2011		
3	Días	45	60					
4								
5								
6	Patente	Marca	Fecha último cambio	Fecha proximo cambio	RETRASO			
7			Filtro Aire	Filtro Aceite	Filtro Aire	Filtro Aceite	Filtro Aire	Filtro Aceite
8	AIS 345	M. Benz	12-01-11	15-01-11				
9	BZX 098	Scania	17-01-11	27-01-11				
10	CFR 271	Scania	22-01-11	01-02-11				
11	DED 340	Scania	27-01-11	06-02-11				
12	CBS 751	M. Benz	01-02-11	11-02-11				
13	CRS 317	M. Benz	06-02-11	16-02-11				
14	AZD 658	Scania	11-02-11	21-02-11				
15	BKJ 109	M. Benz	16-02-11	26-02-11				
16								

Para calcular la fecha del próximo cambio de aire (rango E8:E15) debemos ingresar:

=C8+\$B\$3

Para calcular la fecha del próximo cambio de aceite (rango F8:F15) debemos ingresar:

=D8+\$C\$3

Para calcular la cantidad de días de retraso, en el caso del filtro de aire (rango G8:G15) debemos ingresar:

=F\$2-E8

Para calcular la cantidad de días de retraso, en el caso del filtro de aceite (rango H8:H15) debemos ingresar:

=F\$2-F8

Luego copiamos las fórmulas ingresadas al resto de las celdas.

E8		=C8+\$B\$3						
	A	B	C	D	E	F	G	H
1								
2		Filtro Aire	Filtro Aceite		Fecha Informe	10/05/2011		
3	Días	45	60					
4								
5								
6	Patente	Marca	Fecha último cambio	Fecha próximo cambio	RETRASO			
7			Filtro Aire	Filtro Aceite	Filtro Aire	Filtro Aceite	Filtro Aire	Filtro Aceite
8	AIS 345	M. Benz	12-01-11	15-01-11	26-02-11	16-03-11	73	55
9	BZX 098	Scania	17-01-11	27-01-11	03-03-11	28-03-11	68	43
10	CFR 271	Scania	22-01-11	01-02-11	08-03-11	02-04-11	63	38
11	DED 340	Scania	27-01-11	06-02-11	13-03-11	07-04-11	58	33
12	CBS 751	M. Benz	01-02-11	11-02-11	18-03-11	12-04-11	53	28
13	CRS 317	M. Benz	06-02-11	16-02-11	23-03-11	17-04-11	48	23
14	AZD 658	Scania	11-02-11	21-02-11	28-03-11	22-04-11	43	18
15	BKJ 109	M. Benz	16-02-11	26-02-11	02-04-11	27-04-11	38	13
16								

Para obtener los resultados esperados, el rango E8:F15 debe tener formato Fecha y el rango G8:H15 debe tener formato Número, sin decimales.

Cálculo entre celdas o rangos dispuestos en otra hoja

La presente información se encuentra incluida en la **Hoja1**. En la **Hoja2**, celda **B1**, queremos calcular el total recaudado en el primer bimestre, razón por la cual debemos sumar el contenido de las celdas **B18** y **E18** presentes en la **Hoja1**.

Para realizar dicho cálculo debemos ingresar en la celda B1 (Hoja2) la siguiente fórmula:

=Hoja1!B18 + Hoja1!E18


I26		f_x			
	A	B	C	D	E
1					
2	Mes de Enero 2011		Mes de Febrero 2011		
3	Propietario	Importe a pagar	Propietario	Importe a pagar	
4	María Ovejero	\$ 138,77	María Ovejero	\$ 138,77	
5	Juan Pérez	\$ 333,05	Juan Pérez	\$ 333,05	
6	Sol López	\$ 324,72	Sol López	\$ 324,72	
7	Jorge Fernández	\$ 324,72	Jorge Fernández	\$ 324,72	
8	Mario Sogari	\$ 491,99	Mario Sogari	\$ 491,99	
9	Nélida García	\$ 324,72	Nélida García	\$ 324,72	
10	Lucía Ferraro	\$ 303,91	Lucía Ferraro	\$ 540,00	
11	Carmen López	\$ 641,98	Carmen López	\$ 641,98	
12	Hernán Crespo	\$ 341,37	Hernán Crespo	\$ 341,37	
13	Matías Domingo	\$ 328,88	Matías Domingo	\$ 328,88	
14	Nestor Campos	\$ 341,37	Nestor Campos	\$ 450,00	
15	Isabel Gómez	\$ 341,37	Isabel Gómez	\$ 341,37	
16	Carlos Ferraro	\$ 501,58	Carlos Ferraro	\$ 501,58	
17	Rafael Paz	\$ 458,31	Rafael Paz	\$ 458,31	
18	TOTALES	\$ 5.196,77	TOTALES	\$ 5.541,49	

También podemos realizar dicho cálculo seleccionando las celdas, con el mouse, del siguiente modo:

Ingresamos el signo **=** en la celda **B1** de la **Hoja2**.


Pulsamos sobre la etiqueta de la **Hoja1**, y hacemos clic en **B18**.

En la barra de fórmula visualizaremos:



=Hoja1!B18

Completamos el cálculo agregando, en la barra de fórmulas (sin abandonar la hoja 1), el signo **+** y el contenido de la celda E18, haciendo clic sobre la misma.



=Hoja1!B18+Hoja1!E18

Al pulsar la tecla **ENTER** visualizaremos el resultado de dicha suma en la celda **B1**, Hoja2.


	B1		
		f _x	=Hoja1!B18+Hoja1!E18
	A	B	C
1	Total Enero y Febrero	10738,253	
2			
3			

Para comprender mejor el concepto, en este ejemplo realizaremos un cálculo utilizando celdas dispuestas en tres hojas distintas.

En la celda A5, incluida en la Hoja3 calcularemos

=Hoja1!A2 * Hoja2!B2 + Hoja3!B2


1. Para realizar dicho cálculo, escribimos el signo igual en la celda donde deseamos figure el resultado.




ELEGIR					
	A	B	C	D	E
1					
2	4	2			
3					
4					
5	=				
6					
7					

Hoja1 Hoja2 Hoja3


2. Pulsamos sobre la etiqueta de la hoja que contiene la celda a utilizar.




3. Hacemos clic sobre dicha celda.



4. Agregamos en la barra de fórmulas, de la misma hoja, el operador necesario.



5. Repetimos el procedimiento, a partir del punto 2. con el resto de las celdas, hasta completar el cálculo.



6. Por último pulsamos la tecla **ENTER**.

Uso de operaciones de comparación

Para comparar dos valores se utiliza este tipo de operadores. El resultado de dicha comparación es un valor lógico:

Verdadero o Falso.

En la presente planilla figura en la celda **A3: 5** y en **B3: 9**, utilizando los distintos operadores de comparación obtenemos los siguientes resultados.

	A	B
1		
2	A	B
3	5	9
4		
5	A=B	FALSO
6	A>B	FALSO
7	A<B	VERDADERO
8	A>=B	FALSO
9	A<=B	VERDADERO
10	A<>B	VERDADERO

Operador de comparación	Ejemplo	Valor devuelto (en el ejemplo)
= (Signo igual)	= A3 = B3	Falso
> (Signo mayor que)	= A3 > B3	Falso
< (Signo menor que)	= A3 < B3	Verdadero
>= (Signo mayor o igual que)	= A3 >= B3	Falso
<= (Signo menor o igual que)	= A3 <= B3	Verdadero
<> (Signo distinto que)	= A3 <> B3	Verdadero

Además, los operadores de comparación los utilizamos para establecer la condición que se evalúa en las funciones lógicas, dicha condición debe especificar una expresión que pueda tomar como valor: verdadero o falso. Dicho tema lo abordaremos más adelante, en este mismo capítulo.

Uso de operadores de texto: &

Utilizamos el operador **&** para unir datos de tipo texto. En el siguiente ejemplo vemos que podemos generar la dirección de mail de los empleados de una empresa uniendo el contenido de los campos Nombre (A2) y Apellido (B2) separados por un guion bajo, y el texto común a todos los empleados "@gmail.com". Para ello escribimos en la celda C2:

= A2 & "_" & B2 & "@gmail.com"

	A	B	C	D	E	F
1	Nombre	Apellido	Dirección e- mail			
2	Juan	Perez	Juan_Perez@gmail.com			
3	Sergio	Baez	Sergio_Baez@gmail.com			
4	Viviana	Canosa	Viviana_Canosa@gmail.com			
5	Susana	Gomez	Susana_Gomez@gmail.com			
6	Juan	Calvo	Juan_Calvo@gmail.com			
7						
8						
9						

FUNCIONES EN EXCEL

Microsoft Excel tiene incorporada gran cantidad de funciones divididas en categorías, lo cual nos permite realizar cálculos financieros, manipular fechas y horas, realizar operaciones matemáticas y trigonométricas, como así también calcular valores estadísticos, realizar búsquedas de datos de acuerdo a criterios, manejar bases de datos, etc.

Sintaxis básica

Todas las funciones empiezan con el signo igual seguido por el nombre de la misma y a continuación los parámetros necesarios para dicha función encerrados entre paréntesis. El conjunto de parámetros recibe el nombre de **Argumento de la función**.

=nombre _ función (parámetros)

Para utilizar funciones existen dos métodos:

El primero consiste en ingresar directamente la función en la celda, lo cual


	A	B	C	D	E
1		10%			
2					
3	Precio	Descuento			
4	\$ 120,00	\$ 12,00			
5	\$ 34,00	\$ 3,40			
6	\$ 90,00	\$ 9,00			
7	\$ 89,00	\$ 8,90			
8	\$ 67,00	\$ 6,70			
9	=promedio(
10	PROMEDIO(número1; [número2]; ...)				
11					

requiere el conocimiento previo de la sintaxis correspondiente a la misma. Para ello debemos posicionar el cursor en la celda donde necesitamos insertar la función, escribimos =, luego ingresamos el nombre de la función y por último

los parámetro que ésta requiera, siempre encerrados entre paréntesis.


Al terminar de ingresar el nombre de una función manualmente, Excel nos ofrece, como ayuda, la cantidad de parámetros necesarios para dicha función.

El otro método es utilizando el asistente para funciones, el cual se encuentra en la pestaña **Inicio**, grupo **Modificar**, desplegando el menú **Autosuma**, elegimos la opción **Más funciones....**



Al accionar dicha opción visualizamos la siguiente ventana:

En la misma encontramos un cuadro en donde podremos ingresar qué es lo que deseamos calcular (en caso de no recordar la sintaxis de la función), las funciones clasificadas por categoría, y una breve descripción de la finalidad de la función escogida (la misma se encuentra resaltada en la imagen anterior).



Función SUMA

Como su nombre lo indica, la función **SUMA** se utiliza para calcular el total del conjunto de valores ingresados en el rango especificado.

Sintaxis = SUMA (rango)

Función PROMEDIO

La función promedio calcula el valor medio del conjunto de valores ingresados en el rango especificado.

Sintaxis =PROMEDIO (rango)

Función MÁXIMO

La función **Máximo** devuelve el máximo valor del conjunto de valores ingresados en el rango especificado.

Sintaxis = MAX (rango)

Función MÍNIMO

La función **Mínimo** devuelve el mínimo valor del conjunto de valores ingresados en el rango especificado.

Sintaxis = MIN (rango)

Ejemplo de aplicación de las funciones Suma, Promedio, Máximo y Mínimo:

En la presente planilla se ha consignado los gastos anuales de un comercio, por medio de las siguientes funciones calculamos el gasto total, el promedio de gastos, el gasto más elevado y el menor registrado.

F6		fx		=MAX(B4:B15)			
	A	B	C	D	E	F	G
1	GASTOS MENSUALES						
2	CORRESPONDIENTES AL AÑO 2009						
3	MES	GASTO	GASTO ACUMULADO			ESTADÍSTICAS ANUALES	
4	ENERO	\$ 15.600,00	\$ 15.600,00				
5	FEBRERO	\$ 16.900,00	\$ 32.500,00				
6	MARZO	\$ 12.300,00	\$ 44.800,00			PROMEDIO ANUAL	GASTO MÁXIMO
7	ABRIL	\$ 20.900,00	\$ 65.700,00			\$ 18.347,42	\$ 23.800,00
8	MAYO	\$ 15.789,00	\$ 81.489,00				
9	JUNIO	\$ 21.890,00	\$ 103.379,00				
10	JULIO	\$ 15.790,00	\$ 119.169,00				
11	AGOSTO	\$ 23.800,00	\$ 142.969,00				
12	SEPTIEMBRE	\$ 19.000,00	\$ 161.969,00				
13	OCTUBRE	\$ 21.400,00	\$ 183.369,00				
14	NOVIEMBRE	\$ 18.900,00	\$ 202.269,00				
15	DICIEMBRE	\$ 17.900,00	\$ 220.169,00				
16	TOTAL	\$ 220.169,00					
17							

Para realizar dichos cálculos escribimos:

En la celda B16:	=SUMA (B4:B15)
En la celda E5:	=PROMEDIO (B4:B15)
En la celda F5:	=MAX (B4:B15)
En la celda G5:	=MIN (B4:B15)

Si tuviéramos que consignar los resultados de las funciones precedentes en otra hoja de cálculo, distinta a la que contiene los datos, debemos anteponer al rango que figura en el argumento de la función el nombre de la hoja que contiene dicho rango. Por ejemplo en el caso de la función promedio escribimos:

=PROMEDIO(Hoja1!B4:B15)


Para ejemplificar, supongamos que tenemos distribuida la información anterior del siguiente modo:

En la **Hoja1** los gastos detallados mes a mes y en la **Hoja2**, el gasto promedio, el máximo y el mínimo gasto.


GASTOS MENSUALES CORRESPONDIENTES AL AÑO 2009		
MES	GASTO	GASTO ACUMULADO
1		
2		
3		
4	ENERO	\$ 15.600,00
5	FEBRERO	\$ 16.900,00
6	MARZO	\$ 12.300,00
7	ABRIL	\$ 20.900,00
8	MAYO	\$ 15.789,00
9	JUNIO	\$ 21.890,00
10	JULIO	\$ 15.790,00
11	AGOSTO	\$ 23.800,00
12	SEPTIEMBRE	\$ 19.000,00
13	OCTUBRE	\$ 21.400,00
14	NOVIEMBRE	\$ 18.900,00
15	DICIEMBRE	\$ 17.900,00
16	TOTAL	\$ 220.169,00
17		

ESTADÍSTICAS ANUALES		
PROMEDIO ANUAL	GASTO MÁXIMO	GASTO MÍNIMO
1		
2		
3		
4		
5		
6		
7		
8		
9		
10		
11		
12		
13		
14		
15		
16		
17		


Para ello en la celda Hoja2!A4, escribimos
=Promedio(...y pulsamos sobre la etiqueta de la Hoja1.



Estando en la **Hoja1** seleccionamos, con el mouse, el rango que contiene los gastos mensuales (podemos observar, en la barra de fórmulas, lo ingresado hasta el momento
=promedio(Hoja1!B4:B15.



Por último completamos la fórmula, sin abandonar la **Hoja1**, cerrando el paréntesis en la barra de fórmulas. Finalmente pulsamos la tecla **ENTER**.



Al pulsar la tecla **ENTER**, visualizaremos la **Hoja2** con el resultado devuelto por la función Promedio.

Procediendo de igual modo calculamos el máximo y el mínimo gasto, las fórmulas que debemos escribir son:

Para el máximo gasto: **=Max(Hoja1!B4:B15)**

Para el mínimo gasto: **=Min(Hoja1!B4:B15)**

ESTADÍSTICAS ANUALES		
PROMEDIO ANUAL	GASTO MAXIMO	GASTO MINIMO
\$ 18.347,42	\$ 23.800,00	\$ 12.300,00

También es posible incluir funciones dentro de cálculos. Por ejemplo, si en una determinada celda, por ejemplo **C6**, se desea calcular el margen de temperaturas registradas, estando las mismas ingresadas en el rango **D2:D92**, en dicha celda debemos escribir:

$$= \text{MAX} (\text{D2:D92}) - \text{MIN} (\text{D2:D92})$$

Recordemos que, el margen o amplitud de temperatura es la diferencia existente entre la marca más elevada y la más baja.

Función CONTARA

Sintaxis = CONTARA (rango)

Para aplicar esta función el rango debe contener datos alfanuméricos.

Ejemplo: Para calcular la cantidad de meses consignados en el rango A4:A15 utilizamos:

$$=\text{CONTARA} (\text{A4:A15})$$

GASTOS MENSUALES CORRESPONDIENTES AL AÑO 2009		
MES	GASTO	GASTO ACUMULADO
ENERO	\$ 15.600,00	\$ 15.600,00
FEBRERO	\$ 16.900,00	\$ 32.500,00
MARZO	\$ 12.300,00	\$ 44.800,00
ABRIL	\$ 20.900,00	\$ 65.700,00
MAYO	\$ 15.789,00	\$ 81.489,00
JUNIO	\$ 21.890,00	\$ 103.379,00
JULIO	\$ 15.790,00	\$ 119.169,00
AGOSTO	\$ 23.800,00	\$ 142.969,00
SEPTIEMBRE	\$ 19.000,00	\$ 161.969,00
OCTUBRE	\$ 21.400,00	\$ 183.369,00
NOVIEMBRE	\$ 18.900,00	\$ 202.269,00
DICIEMBRE	\$ 17.900,00	\$ 220.169,00
TOTAL	\$ 220.169,00	

Función CONTAR

Sintaxis = CONTAR (rango)

Para aplicar esta función el rango debe contener datos numéricos.

Ejemplo: Para calcular la cantidad de celdas donde figura los Gastos acumulados utilizamos:

=CONTAR (C4:C15)

GASTOS MENSUALES CORRESPONDIENTES AL AÑO 2009		
MES	GASTO	GASTO ACUMULADO
ENERO	\$ 15.600,00	\$ 15.600,00
FEBRERO	\$ 16.900,00	\$ 32.500,00
MARZO	\$ 12.300,00	\$ 44.800,00
ABRIL	\$ 20.900,00	\$ 65.700,00
MAYO	\$ 15.789,00	\$ 81.489,00
JUNIO	\$ 21.890,00	\$ 103.379,00
JULIO	\$ 15.790,00	\$ 119.169,00
AGOSTO	\$ 23.800,00	\$ 142.969,00
SEPTIEMBRE	\$ 19.000,00	\$ 161.969,00
OCTUBRE	\$ 21.400,00	\$ 183.369,00
NOVIEMBRE	\$ 18.900,00	\$ 202.269,00
DICIEMBRE	\$ 17.900,00	\$ 220.169,00
TOTAL	\$ 220.169,00	

Función CONTAR.BLANCO

Sintaxis: =Contar.blanco (rango)

Utilizamos esta función para contar la cantidad de celdas que no contienen información en un determinado rango.

En la siguiente planilla aplicamos la función Contar.blanco para determinar la cantidad de celdas vacías en el rango destacado.

	A	B
1	7	
2		1
3	9	
4	7	4
5	6	
6	8	3
7		
8	Celdas vacías	6
9		
10		
11		
12		

Funciones lógicas

Función SI

La función **SI** es una función lógica, devuelve un valor determinado si la condición (o criterio) ingresada es verdadera y otro si la condición es falsa.

Sintaxis:

= SI(condición; acción por verdadero; acción por falso)

Para establecer la condición que se evaluará, en las funciones lógicas, utilizamos los operadores de comparación. Dicha condición debe especificar una expresión que pueda tomar el valor lógico verdadero o falso. Los operadores que utilizamos son: =, <, >, <=, >=, <> (siendo la última expresión la que equivale a distinto de...).

Ejemplo:

En la siguiente planilla deseamos calcular la columna Préstamo, teniendo en cuenta el sueldo percibido por un grupo de empleados de la siguiente manera:

	A	B	C	D	E
1	NOMBRE	SUELDO	PRÉSTAMO		
2	Juan	\$ 1.250,00	\$ 10.000,00		
3	María	\$ 2.800,00	\$ 30.000,00		
4	José	\$ 2.500,00	\$ 30.000,00		
5	Ana	\$ 2.000,00	\$ 10.000,00		
6	Carlos	\$ 1.750,00	\$ 10.000,00		
7	Beatriz	\$ 3.000,00	\$ 30.000,00		
8					
9					
10	CONDICIÓN	SUELDO	PRÉSTAMO		
11	menor o igual que	\$ 2.000,00	\$ 10.000,00		
12	mayor que	\$ 2.000,00	\$ 30.000,00		

Si el sueldo es **menor o igual a \$2.000** se le otorgará un préstamo de **\$10.000**, en caso contrario (mayor que \$2.000) el préstamo será de \$30.000.


Por tanto la función ingresada en C2 es:

=SI(B2<= 2000; 10000; 30000).

Luego copiamos dicha fórmula al resto del rango.

Más ejemplos:

En la **columna J** de la siguiente planilla, deseamos que figure el texto **Tour completo** en el caso que la cantidad de reservas (Columna I) sea **superior a 100**, en caso de no cumplirse dicha condición deberá figurar el texto **Promocionar**.



	A	B	I	J
	Cod. Des	DESTINOS	Cantidad de reservas	Condicion de Promocion
1				
2	101	Bariloche	22	Promocionar
3	124	Cordoba	60	Promocionar
4	133	San Martin de los Andes	110	Tour completo
5	154	Las Leñas	100	Promocionar
6	188	Mar del Plata	50	Promocionar
7	195	Tierra del Fuego	120	Tour completo
8	122	Mendoza	20	Promocionar
9	136	Cataratas	260	Tour completo
10	134	Salta- Tres Nubes	220	Tour completo
11	123	Bahía Blanca	90	Promocionar
12	203	Tucumán	65	Promocionar

En J2 ingresamos =SI (I2>100;"Tour Completo";"Promocionar")

Los datos alfanuméricos ingresados en una condición (por ejemplo B2="Tarjeta") o como valor (por ejemplo "Tour Completo") deben encerrarse entre comillas dobles.

Supongamos que, en lugar de pedirnos que si la condición es falsa figure "Promocionar" nos pidiera no escribir nada, en dicho caso deberemos ingresar "", con lo cual quedará la celda en blanco.

Si el resultado de la función forma parte de un cálculo es importante el cero.

Por ejemplo: Supongamos que de acuerdo a las ventas efectuadas por los empleados de una empresa se decide otorgar un premio a aquellos empleados cuyo monto vendido sea mayor o igual a \$10.000, en caso afirmativo dicho premio será una suma fija de \$1.500, en caso contrario no se le asigna cifra alguna.

E4	=SI(C4>=10000;1500;0)				
	A	B	C	D	E
1	Liquidación de Vendedores				
2	Fecha actual		Comisión	10%	5%
3	Apellido	Nombre	Ventas de Mayo	Importe comisión	Premio
4	Gómez	Ana	\$ 7.500,00	\$ 375,00	\$ 0,00
5	Perez	José	\$ 8.900,00	\$ 445,00	\$ 0,00
6	Martinez	Clara	\$ 14.700,00	\$ 1.470,00	\$ 1.500,00
7	Gaspar	Hugo	\$ 7.500,00	\$ 375,00	\$ 0,00
8	Alvarez	Diego	\$ 12.300,00	\$ 1.230,00	\$ 1.500,00
9	Rodriguez	Julian	\$ 9.800,00	\$ 490,00	\$ 0,00
10	Gómez	Ana	\$ 14.700,00	\$ 1.470,00	\$ 1.500,00
11	Perez	José	\$ 7.500,00	\$ 375,00	\$ 0,00
12	Martinez	Clara	\$ 12.300,00	\$ 1.230,00	\$ 1.500,00
13	Martinez	Clara	\$ 7.500,00	\$ 375,00	\$ 0,00
14	Gaspar	Hugo	\$ 8.900,00	\$ 445,00	\$ 0,00
15	Gaspar	Hugo	\$ 14.700,00	\$ 1.470,00	\$ 1.500,00
16	Alvarez	Diego	\$ 23.900,00	\$ 2.390,00	\$ 1.500,00
17	Rodriguez	Julian	\$ 34.000,00	\$ 3.400,00	\$ 1.500,00
18	Gómez	Ana	\$ 2.000,00	\$ 100,00	\$ 0,00
19	Alvarez	Diego	\$ 12.000,00	\$ 1.200,00	\$ 1.500,00
20	TOTALES				

En **E4** ingresamos **=SI(C4>= 10000; 1500; 0)** y luego copiamos dicha fórmula al resto del rango. Seguidamente en **F4** escribimos **= D4 + E4**.

Función Y

Sintaxis =Y (condición_1; condición_2;...; condición_n)

Esta función devuelve como resultado el valor lógico **verdadero** si se cumplen **todas las condiciones simultáneamente**, en caso contrario devuelve el valor lógico falso.

Ejemplo: En la columna C de la siguiente planilla, deberá figurar el valor Verdadero si el empleado pertenece a la categoría A y no faltó, en caso contrario deberá figurar Falso.

La función a ingresar en la celda C3 es:

C3	=Y(A3="A";B3=0)		
	A	B	C
1			
2	Categoría	Asistencia	Premio
3	A	0	VERDADERO
4	A	1	
5	B	0	
6	A	2	
7	B	3	
8	A	0	
9	B	2	
10	A	0	

=Y (A3="A"; B3=0)

Luego de copiar la fórmula ingresada en la celda C3 obtendremos el siguiente resultado:

	A	B	C
1			
2	Categoría	Asistencia	Premio
3	A	0	VERDADERO
4	A	1	FALSO
5	B	0	FALSO
6	A	2	FALSO
7	B	3	FALSO
8	A	0	VERDADERO
9	B	2	FALSO
10	A	0	VERDADERO

Función O

Sintaxis **=O (condición_1; condición_2;...; condición_n)**

Esta función devuelve como resultado el valor lógico **verdadero** si se cumplen **al menos una condición**, en caso contrario devuelve el valor lógico falso.

Ejemplo: En la columna C de la siguiente planilla, deberá figurar el valor Verdadero si el empleado pertenece a la categoría A o no faltó, en caso contrario deberá figurar Falso.

La función a ingresar en la celda C3 es:

=O (A3="A"; B3=0)

Luego de copiar la fórmula ingresada en la celda **C3** obtendremos el siguiente resultado:

	A	B	C
1			
2	Categoría	Asistencia	Premio
3	A	0	VERDADERO
4	A	1	VERDADERO
5	B	0	VERDADERO
6	A	2	VERDADERO
7	B	3	FALSO
8	A	0	VERDADERO
9	B	2	FALSO
10	A	0	VERDADERO

	A	B	C
1			
2	Categoría	Asistencia	Premio
3	A	0	VERDADERO
4	A	1	VERDADERO
5	B	0	VERDADERO
6	A	2	VERDADERO
7	B	3	FALSO
8	A	0	VERDADERO
9	B	2	FALSO
10	A	0	VERDADERO

Función CONTAR.SI

Sintaxis **=CONTAR.SI (RANGO A EVALUAR;"Condición")**

Utilizando la planilla del ejemplo anterior deseamos calcular la cantidad de empleados de la categoría **A**. Como debemos contar teniendo en cuenta una condición (que el empleado pertenezca a la categoría A) debemos utilizar la función **Contar.si**.

	A	B	C	D	E
2	Categoría	Asistencia	Premio		
3	A	0	VERDADERO		
4	A	1	VERDADERO		
5	B	0	VERDADERO		
6	A	2	VERDADERO		
7	B	3	FALSO		
8	A	0	VERDADERO		
9	B	2	FALSO		
10	A	0	VERDADERO		
11					
12	Empleados categoría A		5		
13					
14					

La función ingresada en la celda **C12** es: **=Contar.si (A3:A10;"A")**

Continuando con el ejemplo anterior, en este caso, calcularemos la cantidad de empleados que faltaron más de una vez.

	A	B	C	D	E
2	Categoría	Asistencia	Premio		
3	A	0	VERDADERO		
4	A	1	VERDADERO		
5	B	0	VERDADERO		
6	A	2	VERDADERO		
7	B	3	FALSO		
8	A	0	VERDADERO		
9	B	2	FALSO		
10	A	0	VERDADERO		
11					
12	Empleados categoría A		3		
13					
14					

La función ingresada en la celda C12 es:

=Contar.si (B3:B10;">1")

En este otro ejemplo calcularemos la cantidad de empleados cuyas inasistencias son inferiores o iguales al promedio de inasistencias, figurando dicho promedio en la celda **E3**.

La función ingresada en la celda C12 es:

=Contar.si (B3:B10;"<=" & E3)

	A	B	C	D	E
1					
2	Categoría	Asistencia	Premio		Promedio
3	A	0	VERDADERO		1
4	A	1			
5	B	0			
6	A	2			
7	B	3			
8	A	0			
9	B	2			
10	A	0			
11					
12	Empleados categoría A		5		

Cuando la condición se halla dispuesta en otra celda debemos unir el operador de comparación con dicha celda utilizando el operador de concatenación &.

=Contar.si (B3:B10;"<=" & E3)

Condición dispuesta en la celda E3

En este ejemplo calcularemos, en la siguiente planilla, la cantidad de destinos que comienzan con la letra C.

Como sólo nos interesa la primera letra de los textos ingresados en el campo Destino reemplazaremos las siguientes letras por un comodín.

Podemos utilizar dos tipos de comodines simbolizados por los caracteres * y ?.

Utilizamos un * para representar cualquier serie de caracteres y un ? para representar cualquier carácter individual.

En este caso emplearemos el carácter *, la fórmula a ingresar será:

=Contar.si(B2:B12;"C*")

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C
1			
2	101	Bariloche	22
3	124	Cordoba	60
4	133	San Martin de los Andes	110
5	154	Las Leñas	100
6	188	Mar del Plata	50
7	195	Tierra del Fuego	120
8	122	Mendoza	20
9	136	Cataratas	260
10	134	Salta- Tres Nubes	220
11	123	Bahía Blanca	90
12	203	Tucumán	65
13			
14	Cantidad de tours comienzan con C		2
15			

The formula bar shows: `=CONTAR.SI(B2:B12;"C*")`

Función SUMAR.SI

Sintaxis **=SUMAR.SI (Rango a evaluar;"condición"; rango a sumar)**

Para ejemplificar la presente función calcularemos el monto total abonado en concepto de premios a los empleados de la categoría A.

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E
1					
2	Categoría	Asistencia	Premio		
3	A	0	\$ 2.000,00		
4	A	1	\$ 2.000,00		
5	B	0	\$ 2.000,00		
6	A	2	\$ 2.000,00		
7	B	3	\$ 0,00		
8	A	0	\$ 2.000,00		
9	B	2	\$ 0,00		
10	A	0	\$ 2.000,00		
11					
12	Monto total		\$ 10.000,00		
13					
14					
15					

The formula bar shows: `=SUMAR.SI(A3:A10;"A";C3:C10)`

La función ingresada en la celda E12 es:

=Sumar.si (D3:D10;"A";F3:F10)

Función PI (Opcional)

Sintaxis **=PI ()**

La función PI devuelve el valor 3.141592..... con 15 cifras decimales.

	Radio	Diámetro	Altura	Superficie base	Volumen
Cono 1	3	6	12	28,27	113,10
Cono 2	10	20	30	314,16	3141,59
Cono 3	3	6	4	28,27	37,70
Cono 4	6	12	40	113,10	1507,96

Sup Base = $\pi * R^2$
 Volumen = Sup. Base * Altura / 3

$\pi =$ 3,141592653589790

En la siguiente planilla utilizamos la función PI para calcular el volumen de distintos conos.

En la celda B11 podemos visualizar el valor aproximado de π con 15 cifras decimales.

Función PRODUCTO (Opcional)

Sintaxis **=PRODUCTO (número1; número2; número3;....)**

La función Producto devuelve el producto de los números incluidos en su argumento.

En la siguiente planilla se utiliza la función **Producto** para calcular el costo de fabricar 2500 baldosas rectangulares, cuyas dimensiones son las consignadas en las celdas C3 y D3. La cantidad a producir se halla en A3 y el precio, por cm^2 , en B3.

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F
1	ÁREA CM ²					
2	UNIDADES	VALOR CM ²	BASE	ALTURA	COSTO TOTAL	
3	2500	5	14	8	\$ 1.400.000,00	

The formula bar at the top shows the formula in cell E3: `=PRODUCTO(A3:D3)`.

La fórmula ingresada en E3 es:

=Producto (A3:D3)

Función POTENCIA (Opcional)

Sintaxis **=POTENCIA (base; exponente)**

En la siguiente planilla se calcula el volumen del cilindro teniendo en cuenta el radio de su base y la altura, para lo cual se utilizan la función PI, PRODUCTO y POTENCIA.

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F	G
1	VOLUMEN DEL CILINDRO						
2	RADIO	ALTURA	SUP. BASE	VOLUMEN			
3	2	15	12,57	188,50			
4	4	20	50,27	1005,31			
5	5	25	78,54	1963,50			
6	8	30	201,06	6031,86			

The formula bar at the top shows the formula in cell C3: `=PI()*POTENCIA(A3;2)`.

La fórmula ingresada en C3 es:

=Pi() * Potencia (A3; 2)

La fórmula ingresada en D3 es:

=Producto (B3; C3)

Función RAIZ (Opcional)

Sintaxis =RAIZ (NÚMERO)

La función **Raiz** devuelve como resultado la **raíz cuadrada** del número ingresado en el argumento de la misma.

	A	B	C	D
1	NÚMERO	RAÍZ CUADRADA		
2	1	1		
3	4	2		
4	9	3		
5	16	4		
6	25	5		
7	36	6		
8	49	7		
9	64	8		
10	81	9		
11	100	10		

Función NO (Opcional)

Sintaxis =No (celda)

Esta función devuelve el valor lógico opuesto al valor lógico de la celda que figura en su argumento.

Ejemplo:

	A	B	C
1	VALOR1	VALOR2	
2	VERDADERO	FALSO	
3	FALSO	VERDADERO	
4			
5			
6			
7			

Funciones de fecha y hora

Función HOY

Sintaxis =HOY ()

La función **Hoy** devuelve como resultado la fecha actual con formato de fecha.

Función AÑO

Sintaxis =Año(fecha)

La función **Año** devuelve el año, un número entero entre 1900 y 9999.

Función MES

Sintaxis =Mes(fecha)

La función **Mes** devuelve el mes, un número entero entre 1 (Enero) y 12 (Diciembre).

Función SIFECHA

La función **Sifecha** devuelve la diferencia entre dos fechas, expresada en el tiempo de información que se desea obtener.

Sintaxis =SIFECHA(fecha_inicial; fecha_final; unidad)

Siendo fecha_inicial menor (más antigua) que fecha_final.

El argumento unidad es el tiempo de información en el que Excel devolverá el resultado. Puede ser uno de los siguientes valores:

"m" - meses. Número de meses completos entre fecha_inicial y fecha_final.

"d" - días. Número de días entre fecha_inicial y fecha_final.

"y" - años. Número de años completos entre fecha_inicial y fecha_final.

"ym" - meses excluyendo años.

"yd" - días excluyendo años.

"md" - días excluyendo meses y años.

Función FECHA

Sintaxis =Fecha (Año; Mes; Día)

La función Fecha devuelve como resultado el número que representa la fecha en código de fecha y hora de Microsoft Excel.

Por ejemplo la función =FECHA(2010;4;30) devuelve como resultado 30/04/2010.

Función AHORA

Sintaxis = Ahora ()

Devuelve la fecha y hora actuales con formato de fecha y hora.

Función DIASEM

Sintaxis = DIASEM (FECHA; TIPO)

Devuelve un número entre 1 y 7, que identifica el día de la semana. Si ingresamos en tipo el valor 1, el domingo estará representado por 1, el lunes por 2,... y por último el sábado por 7.

Ejemplo de aplicación de las funciones fecha y hora:

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F
1	FECHA ACTUAL	10/05/2011		Hora	10/05/2011 23:34	
2						
3						
4						
5	LÓPEZ	20/01/1998	13	3	21	
6	PEREZ	23/03/2000	11	1	18	
7	SANCHEZ	05/06/2002	8	11	5	
8	BAEZ	12/03/2002	9	1	29	
9	FERNÁNDEZ	17/11/2005	5	5	24	
10	ALVAREZ	12/08/1985	25	8	29	
11						
12						

The formula bar at the top shows: =SIFECHA(B5;\$B\$1;"Y")

Las funciones utilizadas son:

En B1: =HOY()

En E1: =AHORA()

En C5: =SIFECHA(B5;\$B\$1;"y")

En D5: =SIFECHA(B5;\$B\$1;"ym")

En E5: =SIFECHA(B5;\$B\$1;"md")

Luego copiamos al resto del rango.

Funciones matemáticas:

Función ENTERO

Sintaxis =Entero (número)

La función **Entero** redondea el número que figura en el argumento al entero inmediato anterior.

Ejemplo: =Entero (9,7) devuelve el valor 9.

=Entero (-1,5) devuelve el valor -2

Función REDONDEAR

Sintaxis =Redondear (número; núm_decimales)

La función **Redondear** aproxima el número contenido en el primer argumento de la función con la cantidad de decimales que se indica en el segundo argumento (*núm_decimales*). Para redondear, Excel utiliza el siguiente procedimiento (al igual que en Matemática):

Por ejemplo, si el segundo argumento, *núm_decimales*, es 0 observa la cifra decimal. Si ésta es mayor o igual a 5 le suma 1 a la parte restante. De lo contrario elimina las cifras decimales.

=Redondear(4,5;0) devuelve 5

=Redondear(-4,5;5) devuelve -5

Función TRUNCAR

Sintaxis =Truncar (número; núm_decimales)

La función **Truncar** aproxima el número contenido en el primer argumento con la cantidad de decimales que se indica en el segundo argumento, sin importar qué valor que valor tiene la siguiente cifra decimal.

En la siguiente planilla podemos observar el resultado de aplicar las funciones Entero, Redondear y Truncar a los números contenidos en el rango A3:A5.

	A	B	C	D
1				
2	Número	Entero	Truncar	Redondear
3	34,1249	34	34,12	34,12
4	34,1889	34	34,18	34,19
5	34,9987	34	34,99	35

Número	Formato	FUNCIÓN		
		ENTERO	TRUNCAR	REDONDEAR
1,5	2	1	1	2
-1,5	-2	-2	-1	-2
4,2	4	4	4	4
-4,2	-4	-5	-4	-4
7,8	8	7	7	8
-7,8	-8	-8	-7	-8
2,069	2	2	2	2
-2,069	-2	-3	-2	-2
5,061	5	5	5	5
-5,061	-5	-6	-5	-5

Comparemos estas funciones aplicadas a valores negativos e incluyamos los efectos de eliminar decimales a través de Formato.

Funciones de búsqueda y referencia:

Función BUSCARV

Sintaxis =BuscarV (valor_buscado; matriz_de_búsqueda; columna; 0 ó 1)

La función **BuscarV** busca en la primera columna de una tabla o matriz de valores y devuelve el valor en la misma fila desde una columna especificada.

Dada una matriz o rango en donde figura el número de legajo, el nombre y el apellido de una lista de empleados, deseamos ingresar una función en las celdas **B11** y **C11** de modo tal que, al ingresar el número de legajo en la celda **A11**, aparezca automáticamente el nombre y apellido del empleado.

Como el valor por el cual se buscará es el número de legajo, la columna que contiene dicha información deberá figurar a la izquierda de los valores a buscar.

Utilizaremos, por lo tanto, la función **BuscarV**:

=BuscarV (A10;A2:C7;2;0)

	A	B	C	D	E
1	LEGAJO	NOMBRE	APELLIDO		
2	1	JUAN	PÉREZ		
3	2	PEDRO	GARCÍA		
4	3	ANIBAL	FERNÁNDEZ		
5	4	ROSA	MOSQUERA		
6	5	MARIA	RIZZO		
7	6	JUANA	MOLINA		
8					
9	LEGAJO	NOMBRE	APELLIDO		
10	2	PEDRO	GARCÍA		
11					

En **Valor_buscado** ingresamos **A10**, ya que en dicha celda escribiremos el número de legajo.

En **Matriz_de_búsqueda** ingresamos el rango A2:C7.

Como **Indicador de columna** y en caso de querer obtener el nombre del empleado ingresamos: 2 (segunda columna), para el apellido ingresamos: 3 (tercer columna)

El cuarto argumento indica si la búsqueda tiene que ser por valor exacto (colocamos 0), o puede encontrar el valor más aproximado (colocamos 1).

Para trabajar con valores aproximados la lista debe estar ordenada en forma ascendente.

Veamos el siguiente ejemplo, en el cual deseamos asignar a cada empleado el porcentaje correspondiente de acuerdo a su antigüedad, dicha información se encuentra incluida en el rango B3:C8.

La función ingresada en E11 es:

=BuscarV(D11;\$B\$3:\$C\$8;2;1)

Como el cuarto argumento es 1, si no encuentra el valor especificado devolverá el valor inmediato anterior. Por ejemplo, si la antigüedad a buscar es 6 años, al no encontrar en forma exacta dicho valor, devolverá el valor 5 (inmediato inferior).

The screenshot shows an Excel spreadsheet with the following data:

Porcentaje según antigüedad	
Años	Porcentaje
0	0%
5	10%
10	20%
15	30%
20	35%
25	50%

Código	Empleado	Sueldo básico	Antigüedad	Porcentaje
1	López Augusto	1000	1	0%
2	Suárez Alberto	1200	6	10%
3	Martínez Emilio	1100	12	20%
4	Gutiérrez Ana	1200	25	50%
5	Méndez, Gabriel	1000	20	35%
6	López, Ana	1300	35	50%

The formula bar shows: `=BUSCARV(D12;B3:C8;2;1)`. The cell E12 contains the result 10%.

Resumiendo, los cuatro argumentos de las Funciones BuscarV y BuscarH pueden pensarse como:

(¿qué busco?; ¿dónde busco?; ¿qué muestro?; ¿exacto o aproximado?)

(Valor_buscado; matriz_de_búsqueda; indicador; 1 ó 0)

Función BUSCARH

Sintaxis **=BuscarH (valor_buscado; matriz_buscar_en; fila ; 0 ó 1)**

La función BuscarH busca en la primera fila de una tabla o matriz de valores y devuelve el valor en la misma columna desde una fila especificada.

En la siguiente planilla utilizamos la función **Buscarh** para completar la columna **E (Porcentaje descuento)**, teniendo en cuenta la información contenida en el rango **B2:F3**. La función ingresada en **E6** es:

=Buscarh (C6; \$B\$2:\$F\$3;2;0)

E6 fx =BUSCARH(C6;\$B\$2:\$F\$3;2;0)

	A	B	C	D	E	F
1	Porcentaje de descuento					
2	Rubro	CERAMICA	DEPORTES	HISTORIA	MANUALIDADES	TURISMO
3	Porcentaje	10%	5%	20%	10%	15%
4						
5	Título	Autor	Rubro	Precio Unitario	Porcentaje descuento	
6	AGRICULTURA EUROPA OCCIDENT MODERNA	ARDIT	HISTORIA	\$ 22,00	20%	
7	AMERICA LATINA Y COLONIALISMO EUROPA	PEREZ	HISTORIA	\$ 24,00	20%	
8	ATLAS ANTIGUO DE AMERICA S.XV Y XVI	VARGAS	HISTORIA	\$ 16,00	20%	
9	CERAMICA FINA	NORTON	CERAMICA	\$ 68,00	10%	
10	CERAMICA VIVA	CARUSO	CERAMICA	\$ 76,00	10%	
11	CERAMICA VIVA	CARUSO	CERAMICA	\$ 76,00	10%	
12	CERAMICA Y HORNOS MODERNOS	DE LA POZA	CERAMICA	\$ 20,00	10%	
13	CERAMICA, PASTAS Y VIDRIADOS	VITTEL	CERAMICA	\$ 30,00	10%	
14	CERAMICA, PASTAS Y VIDRIADOS	VITTEL	CERAMICA	\$ 30,00	10%	
15	COMP HISTORIA REVOLUC FRANCESA	SOBOUL	HISTORIA	\$ 29,00	20%	
16	DEL ESCRIBANO A LA BIBLIOTECA	BOUZA	HISTORIA	\$ 31,00	20%	
17	ECONOMIA EUROPEA EN EL SIGLO XVI	ALVAR	HISTORIA	\$ 24,00	20%	
18	EJERCICIOS PARA LA TERCERA EDAD	LOUWARD	DEPORTES	\$ 32,00	5%	
19	EL CALENTAMIENTO EN EL DEPORTE	FREWALD	DEPORTES	\$ 7,00	5%	
20	EL COLONIALISMO (1815-1873)	HERNANDEZ	HISTORIA	\$ 21,00	20%	
21	ELIMINA BARRIGA (THE CRUNCH)	AMEN	DEPORTES	\$ 78,00	5%	
22	EN FORMA DESPUES DE LOS 40	RODRIGUEZ	DEPORTES	\$ 10,00	5%	
23	FORMULARIO Y PRACTICAS DE CERAMICA	ARTIGAS	CERAMICA	\$ 43,00	10%	
24	FORMULARIO Y PRACTICAS DE CERAMICA	ARTIGAS	CERAMICA	\$ 43,00	10%	
25	GUERR. NAPOLEONICAS HASTA REV 1848	RUDE	HISTORIA	\$ 20,00	20%	
26	HIST DE LA REVOL FRANCESA	SCHMITT	HISTORIA	\$ 17,00	20%	

Hoja1 Hoja2 Hoja3 100% +

Función ELEGIR

Sintaxis =Elegir (número_índice; valor1; valor2;...)

La función **Elegir** devuelve un valor de la lista de argumentos de acuerdo al número índice ingresado.

Por ejemplo la función =Elegir (4; "Enero"; "Febrero"; "Marzo"; "Abril"; "Mayo"; "Junio"; "Julio"; "Agosto"; "Septiembre"; "Octubre"; "Noviembre"; "Diciembre") devuelve Abril por figurar dicho valor en cuarto lugar.

	A	B
1	Num_índice	Mes correspondiente
2	4	Abril
3	8	Agosto
4	3	Marzo
5	1	Enero
6	2	Febrero
7	6	Junio
8	7	Julio
9	10	Octubre
10	9	Setiembre
11	5	Mayo
12	11	Noviembre
13	12	Diciembre
14		

Si dicha función la aplicamos en la presente planilla debemos reemplazar el número índice por la dirección de la celda que lo contiene, en este caso A2.

La fórmula ingresada para hallar los valores en la segunda columna es:

=ELEGIR(A2; "Enero"; "Febrero"; "Marzo"; "Abril"; "Mayo"; "Junio"; "Julio"; "Agosto"; "Setiembre"; "Octubre"; "Noviembre"; "Diciembre")

Funciones de texto

Función CONCATENAR

Sintaxis =CONCATENAR (texto1; texto2; texto3;...)

La función concatenar la utilizamos para unir, en una única celda, el contenido de dos o más celdas. También podemos emplear el operador **&**.

A10	=CONCATENAR(D5;" ";A5)				
	A	B	C	D	E
1	Mercadería	Existencia	Mínimo requerido	Estado	
2	Resma papel A4	34	50	Reponer	
3	Resma papel Carta	60	50	No reponer	
4	Cajas de sobres tamaño carta	45	70	Reponer	
5	Bolígrafos por 30	60	60	No reponer	
6					
7	Reponer Resma papel A4				
8	No reponer Resma papel Carta				
9	Reponer Cajas de sobres tamaño carta				
10	No reponer Bolígrafos por 30				
11					

Empleando el operador **&** debemos ingresar en A10: **= D5 & " "& A5**

El argumento de la función puede ser una referencia de celda, o un texto entre comillas dobles.


En la presente planilla hemos generado la dirección de mail de un conjunto de empleados de una empresa.

C3	=CONCATENAR(A3;"_";B3;"@gmail.com")				
	A	B	C	D	E
1	Datos del empleado				
2	Nombre	Apellido	eMail		
3	Juan	Pérez	Juan_Pérez@gmail.com		
4	María	López	María_López@gmail.com		
5	Sol	Solano	Sol_Solano@gmail.com		
6	Mariela	Gómez	Mariela_Gómez@gmail.com		
7	Gabriel	González	Gabriel_González@gmail.com		

Función Nompropio (Opcional)

Sintaxis: =Nompropio (texto)

La función Nompropio convierte en mayúscula la primera letra de cada palabra.



Función DERECHA

Sintaxis: =Derecha(texto; num_caracteres)

La función Derecha devuelve como resultado el número de caracteres especificado del principio de una cadena de texto.

Función IZQUIERDA

Sintaxis: =Izquierda(texto; num_caracteres)

La función Izquierda devuelve como resultado el número de caracteres especificado desde el final de una cadena de texto.

Función ESPACIOS

Sintaxis: =Espacios(texto)

La función **Espacios** quita todos los espacios en blanco que figuran junto al texto, excepto los espacios individuales entre palabras.

Función EXTRAE

Sintaxis =EXTRAE (texto; posición_inicial; número de caracteres)

La función **Extrae** devuelve como resultado los caracteres de una cadena de texto, dada la posición inicial y la cantidad de caracteres.

En la celda B39 utilizamos: **=EXTRAE (A39; 1; 1)**, en C39: **=EXTRAE (A39; 1; 3)** y en D39: **=EXTRAE (A39; 4; 3)**.

	D39			fx	=EXTRAE(A39;4;3)
		A	B	C	D
37					
38					
39	Murcielago	M	Mur		cie
40					
41					

Función LARGO

Sintaxis =Largo (Texto)

La función Largo devuelve como resultado el número de caracteres de un texto.

Ejemplo de aplicación de funciones de texto.

	C15					
		A	B	C	D	E
1						
2		Texto				Funciones de texto
3			Izquierda (A5;3)	Derecha(A5;3)	Extrae(A4;3;4)	Largo(A5)
4	Espacio	Publicitario	Esp	rio	paci	26
5						Espacio Publicitario

Función HALLAR

Sintaxis =Hallar(texto_buscado; dentro_del_texto; núm_inicial)

Busca un texto dentro de otro y devuelve la posición del texto buscado. Realiza la búsqueda leyendo el texto de izquierda a derecha a partir de la posición inicial indicada en núm_inicial. A diferencia de la función ENCONTRAR, HALLAR no distingue entre mayúsculas y minúsculas y admite caracteres comodines (? un solo carácter, * cualquier número de caracteres).

Ejemplo: Supongamos que tenemos la siguiente lista de empleados de donde se extraen los datos para el envío anual de correspondencia no formal (salutación de cumpleaños, Fin de año, etc.). Para éste tipo de correspon-

dencia, se estila escribir primero el Nombre (en **F**) y luego el Apellido (en **G**).

	A	B	C	D	E	F	G
	Identificación	Apellido y Nombre	Cód. Categoría	Fecha Último Examen	Fecha Próximo Examen	Nombre	Apellido
1	TCP-AGUMA	AGUDO, FATIMA	TCP	15/11/2007	15/11/2008	FATIMA	AGUDO
2	TCP-ALAI	ALARZA, DAVID	TCP	22/10/2007	22/10/2008	DAVID	ALARZA
3	MEC-ALCLA	ALCORTA, PAULA	MEC	28/09/2007	28/09/2009	PAULA	ALCORTA
4	PIL-ALORO	ALONSO, ALVARO	PIL	04/09/2007	04/03/2008	ALVARO	ALONSO
5	TCP-ALVRA	ALVAREZ, SARA	TCP	01/05/2007	01/05/2008	SARA	ALVAREZ
6	TCP-ARESA	ARENAS, VANESA	TCP	18/08/2007	18/08/2008	VANESA	ARENAS

=Hallar(" "; B2) encontraría la posición donde se encuentra el espacio en blanco que separa el apellido del nombre (también se podría buscar ","). Se obtendrían los siguientes valores:

	B	C	D	E	F	G	H
	Apellido y Nombre	Cód. Categoría	Fecha Último Examen	Fecha Próximo Examen	Nombre	Apellido	
	AGUDO, FATIMA	TCP	15/11/2007	15/11/2008	FATIMA	AGUDO	7
	ALARZA, DAVID	TCP	22/10/2007	22/10/2008	DAVID	ALARZA	8
	ALCORTA, PAULA	MEC	28/09/2007	28/09/2009	PAULA	ALCORTA	9
	ALONSO, ALVARO	PIL	04/09/2007	04/03/2008	ALVARO	ALONSO	8
	ALVAREZ, SARA	TCP	01/05/2007	01/05/2008	SARA	ALVAREZ	9
	ARENAS, VANESA	TCP	18/08/2007	18/08/2008	VANESA	ARENAS	8

Aprovecharíamos la Función Largo para obtener la longitud de la cadena:

	B	C	D	E	F	G	H	I
	Apellido y Nombre	Cód. Categoría	Fecha Último Examen	Fecha Próximo Examen	Nombre	Apellido		
	AGUDO, FATIMA	TCP	15/11/2007	15/11/2008	FATIMA	AGUDO	7	13
	ALARZA, DAVID	TCP	22/10/2007	22/10/2008	DAVID	ALARZA	8	13
	ALCORTA, PAULA	MEC	28/09/2007	28/09/2009	PAULA	ALCORTA	9	14
	ALONSO, ALVARO	PIL	04/09/2007	04/03/2008	ALVARO	ALONSO	8	14
	ALVAREZ, SARA	TCP	01/05/2007	01/05/2008	SARA	ALVAREZ	9	13
	ARENAS, VANESA	TCP	18/08/2007	18/08/2008	VANESA	ARENAS	8	14

=Largo(B2)

Nos queda ahora razonar y combinar estos resultados con Derecha e Izquierda para desarmar los datos de la columna **B** acorde a lo necesitado. Si al largo total le restamos la el número de posición del " ", tendremos la longitud del nombre del empleado:

=Largo(B2) - Hallar(" ";B2)

Aplicamos las Funciones de lateralidad en cadenas para obtener finalmente, el Nombre en **F** y el Apellido en **G**:

=Derecha(B2; Largo(B2) – Hallar(" ";B2))


=Izquierda(B2; Hallar(" ";B2) – 2)

En la última Función restamos 2 para descontar " ," (el espacio y la coma).

The first screenshot shows the Excel interface with the formula bar displaying `=DERECHA(B2;LARGO(B2)-HALLAR(" ";B2))` in cell F2. The second screenshot shows the formula bar displaying `=IZQUIERDA(B2;HALLAR(" ";B2)-2)` in cell G2.

Identificación	Apellido y Nombre	Cód. Categoría	Fecha Último Examen	Fecha Próximo Examen	Nombre	Apellido
2 TCP-AGUMA	AGUDO, FATIMA	TCP	15/11/2007	15/11/2008	FATIMA	AGUDO
3 TCP-ALAD	ALARZA, DAVID	TCP	22/10/2007	22/10/2008	DAVID	ALARZA
4 MEC-ALCLA	ALCORTA, PAULA	MEC	28/09/2007	28/09/2008	PAULA	ALCORTA
5 PIL-ALORO	ALONSO, ALVARO	PIL	04/09/2007	04/09/2008	ALVARO	ALONSO
6 TCP-ALVRA	ALVAREZ, SARA	TCP	01/05/2007	01/05/2008	SARA	ALVAREZ
7 TCP-ARESA	ARENAS, VANESA	TCP	18/08/2007	18/08/2008	VANESA	ARENAS

Análisis visual:



Funciones de información

Función ESERROR

Sintaxis: **=eserror(referencia)**

La utilizamos para manejar los errores que puedan producirse en Excel al momento de trabajar con fórmulas.

A continuación encontramos una lista con los errores frecuentes que obtenemos en Excel, junto a su significado.

Error	Significado
#iDIV/0!	Se produce al hacer una división por cero, o bien, por una referencia a un cero.
#iVALOR!	Similar a #iNUMERO!, lo obtenemos cuando el tipo de argumento solicitado por la función, es distinto al ingresado por el usuario.
#iNULO!	Una fórmula da este error cuando la celda de intersección de dos rangos, no existe. En Excel, el operador de intersección es un espacio en blanco. Por tanto, la fórmula =A2:D2 J1:J10, devuelve #iNULO! ya que los rangos A2:D2 y J1:J10 no se intersectan en ninguna celda.
#iN/A	Se produce cuando una fórmula de búsqueda o referencia no encuentra ninguna coincidencia exacta en la correspondiente matriz de búsqueda. Significa que el valor buscado no existe en la matriz de búsqueda
#iREF!	Este tipo de error surge cuando tenemos una referencia de celda inválida en la fórmula. Por ejemplo, cuando se intenta buscar en la tercera columna de una matriz que solo tiene dos columnas.
#iNUMERO!	Este error se produce cuando ingresamos algún valor no numérico como un argumento de función, que Excel espera que sea argumento numérico (o una referencia a un valor numérico). Otra posibilidad es ingresar un número inválido, como uno negativo cuando se espera uno positivo, o un 2 cuando el argumento solo admite 0 ó 1.
#iNOMBRE!	Este error lo obtenemos cuando escribimos mal el nombre de alguna función.

La función **ESERROR** devuelve el valor lógico Verdadero si el valor contenido en la celda es erróneo y el valor lógico falso en caso contrario. En la siguiente planilla se ha ingresado en la celda **A12** la siguiente fórmula:

=SI (ESERROR (B10); "Nombre inexistente"; "")

	A	B	C	D	E	F	G
1	LEGAJO	NOMBRE	APELLIDO				
2	1 JUAN	PÉREZ					
3	2 PEDRO	GARCÍA					
4	3 ANIBAL	FERNÁNDEZ					
5	4 ROSA	MOSQUERA					
6	5 MARIA	RIZZO					
7	6 JUANA	MOLINA					
8							
9	LEGAJO	NOMBRE	APELLIDO				
10		#N/A	#N/A				
11							
12	Nombre inexistente						

Función ESBLANCO

Sintaxis: **=esblanco(referencia)**

Nos informa si la celda a la que hace referencia esta función está vacía (sin valor alguno). Devolverá Verdadero en caso de estar en blanco y Falso en caso de contener algún tipo de valor.

Atención: al utilizar esta función hay que tener en cuenta que si tenemos una celda en la cual se ha tipeado un espacio (con la barra espaciadora), dicha celda en apariencia parecerá vacía, no obstante el resultado de aplicar la función ESBLANCO dará falso, dado que un espacio es en si mismo un valor.

Función ESNUMERO

Sintaxis: **=esnumero(referencia)**

Nos informa si la celda a la que hace referencia esta función contiene un número. Devolverá Verdadero en caso de contener un valor numérico (recordar que las fechas son tratadas como números) y Falso en caso de contener otro tipo de valor.

Función JERARQUÍA

Permite conocer qué posición ocupa, por su tamaño, un valor numérico dentro de una lista.

Sintaxis: **=Jerarquia(número; referencia;[orden])**

Número: es el número cuya jerarquía (clasificación) deseamos conocer.

Referencia: es la referencia del rango que contiene la lista de valores. Los valores no numéricos se pasan por alto.

Orden: es un número que especifica cómo clasificar el argumento número. Si ingresamos 0 se calcula la posición que ocuparía el número si la lista es-

tuviese ordenada de mayor a menor, si ingresamos 1 se calcula la posición que ocuparía el número si la lista estuviese ordenada de menor a mayor.

	C2			f_x	=JERARQUIA(B2:\$B\$2:\$B\$6;0)
	A	B	C	D	
1	Nombre	Puntaje	Orden de Mérito		
2	Juan	65		4	
3	Pedro	29		5	
4	María	90		2	
5	José	78		3	
6	Elena	100		1	
7					

Función K.ESIMO.MAYOR

Permite obtener, dentro de una lista de valores numéricos, el que ocupa una posición determinada contando a partir del mayor.

Sintaxis **=K.esimo.mayor(matriz;k)**

Matriz: Es el rango de datos cuyo k-ésimo mayor valor deseamos determinar.

K representa la posición (a partir del mayor valor), dentro del rango de celdas de los datos que se van a devolver.

Por ejemplo si ingresamos en la celda D2:

=K.esimo.mayor(\$B\$2:\$B\$6;C2),

el valor devuelto (**65**) corresponde al valor que ocupa la cuarta posición cuando ordenamos la lista de mayor a menor.

	D2			f_x	=K.ESIMO.MAYOR(\$B\$2:\$B\$6;C2)
	A	B	C	D	
1	Nombre	Puntaje	Orden de Mérito	K.esimo.Mayor	
2	Juan	65	4		65
3	Pedro	29	5		29
4	María	90	2		90
5	José	78	3		78
6	Elena	100	1		100
7					

Función K.ESIMO.MENOR

Permite obtener, dentro de una lista de valores numéricos, el que ocupa una posición determinada contando a partir del menor.

Sintaxis **=K.esimo.menor(matriz;k)**

Matriz es el rango de datos numéricos cuyo k-ésimo menor valor deseamos determinar.

K es la posición, dentro del rango de celdas, de los datos que se van a devolver.

Por ejemplo si ingresamos en la celda D2:

=K.esimo.menor(\$B\$2:\$B\$6;C2)

el valor devuelto (**90**) corresponde al valor que ocupa la cuarta posición cuando ordenamos la lista de menor a mayor.

Podemos observar, efectivamente que, cuando la lista está ordenada en orden decreciente el valor que ocupa la cuarta posición es 65 en cambio, si la lista se encuentra ordenada en orden creciente el valor que ocupa la posición 4 es 90.

Lista ordenada de mayor a menor				Lista ordenada de menor a mayor
100				29
90				65
78				78
65				90
29				100

Cuarta posición

Función SUMAR.SI.CONJUNTO

Permite sumar valores que cumplan con **varios criterios** simultáneamente.

Sintaxis **=Sumar.si.conjunto(rango_suma; rango_criterios1; criterio1; [rango_criterios2; criterio2]; ...)**

A partir del rango_criterios2, es opcional. Se permiten hasta 127 pares de rango_criterios; criterio.

Ejemplo: De la siguiente planilla con Montos de Ejecución de Obras para distintos organismos y diferentes fuentes de financiamiento, se desea obtener los Montos finales correspondientes a ciertas Unidades Académicas, cuyo estado sea "Terminado".

	A	B	C	D	E	F
	Obra	Unidad Académica	Fuente de financiamiento	Monto(en pesos)	Estado	Plazo de realización (en días)
4						
5	Nueva Alimentación Eléctrica	Facultad de Bellas Artes	CONICET	\$ 48.832	En proceso licitatorio	3
6	Remodelación Baños subsuelo	Facultad de Humanidades	Ministerio de Educación de la Nación	\$ 56.811	En proceso licitatorio	4
7	Aula Entrepiso	Facultad de Humanidades	Presupuesto UNLP	\$ 388.000	Para adjudicar	4
8	Salida de Emergencia	Bachillerato de Bellas Artes	Gobierno Pcia. Bz. As.	\$ 6.500	Terminado	5
9	Reparación Ascensores	Facultad de Cs. Jurídicas	Ministerio de Planificación de la Nación	\$ 180.000	Terminado	12
10	Reparación Dugradiente de Homón en escuela	Facultad de Humanidades	Recursos de la Facultad	\$ 4.230	Terminado	
11	Escalera Principal Edificio Píeforma Universitaria	Facultad de Cs. Jurídicas	Ministerio de Educación de la Nación	\$ 10.890	Terminado	2
12	Equipamiento sala de Grabados	Facultad de Bellas Artes	Presupuesto UNLP	\$ 86.270	Para adjudicar	6
13	Construcción Aulas Planta Baja	Facultad de Cs. Económicas	Recursos de la Facultad	\$ 91.996	En ejecución	12
14	Nuevo Edificio IALP	Facultad de Cs. Astronómicas	CONICET	\$ 639.770	Terminado	24
15	Acceso y Ascensor discapacitados	Facultad de Bellas Artes	Ministerio de Educación de la Nación	\$ 59.197	Para adjudicar	9
16	Reparación Aulas	Facultad de Bellas Artes	Presupuesto UNLP	\$ 143.500	Terminado	9
17	Construcción Aulas 2º Piso	Facultad de Cs. Económicas	Recursos de la Facultad	\$ 58.470	En ejecución	18
18	Nuevo Edificio IARI	Facultad de Cs. Astronómicas	CONICET	\$ 1.256.282	Terminado	53
19	Remodelación Entrepiso	Facultad de Cs. Jurídicas	Ministerio de Educación de la Nación	\$ 24.860	Para adjudicar	2
20	Sanitarios Para discapacitados	Facultad de Bellas Artes	Presupuesto UNLP	\$ 25.277	Terminado	6
21	Casa de Postgrado	Facultad de Cs. Jurídicas	Recursos de la Facultad	\$ 411.975	Terminado	16
22	Museo de Cs. Naturales Restauración de Fachada	Facultad de Cs. Naturales y Medio	Gobierno Pcia. Bz. As.	\$ 182.412	Terminado	9
23	Reparación Canaletas	Facultad de Cs. Exactas	Ministerio de Planificación de la Nación	\$ 12.144	Terminado	6
24	Remodelación Baños 1º y 2º piso	Facultad de Humanidades	Presupuesto UNLP	\$ 39.500	Para adjudicar	6
25	Museo de Cs. Naturales Tratamiento Humedad	Facultad de Cs. Naturales y Medio	Gobierno Pcia. Bz. As.	\$ 159.800	Terminado	9
26	Renovación Cubiertas y Cielosazos	Facultad de Cs. Económicas	Ministerio de Planificación de la Nación	\$ 22.830	Para adjudicar	3
27	Escalera de Emergencia	Facultad de Humanidades	Presupuesto UNLP	\$ 190.096	En ejecución	3
28	Despachos Instituto de Física	Facultad de Cs. Exactas	CONICET	\$ 370.000	En ejecución	18
29	Construcción Grupo Sanitario	Facultad de Cs. Jurídicas	Ministerio de Educación de la Nación	\$ 14.950	En proceso licitatorio	3
30	Adecuación Instalación de gas	Facultad de Cs. Jurídicas	Presupuesto UNLP	\$ 6.500	En proceso licitatorio	6
31	Edificio Píeforma Universitaria Etapa I	Facultad de Cs. Jurídicas	CONICET	\$ 162.350	Para adjudicar	18
32	Adecuación Área Académica	Facultad de Cs. Jurídicas	Ministerio de Planificación de la Nación	\$ 262.527	En ejecución	12

41	Liceo Víctor Mercante Restauración Etapal	Liceo V. Mercante	Presic
42	Aula para conferencias	Facultad de Trabajo Social	Recur
43			
44			
45			
46			
47			
48	Unidad Académica	Terminado	
49	Facultad de Bellas Artes	\$ 174.877	
50	Facultad de Humanidades	\$ 4.230	
51	Facultad de Cs. Jurídicas	\$ 612.565	
52	Facultad de Cs. Astronómicas	\$ 1.896.002	
53		\$ 2.687.674	
54			
55			
56			

Aquí se hace necesario especificar **dos** criterios para seleccionar los valores parciales a sumar y obtener el Monto total:

- ☞ Unidad Académica
- ☞ Estado "Terminado"

Aplicando esta función será:

=Sumar.si.conjunto(D5:D42;B5:B42;A49;E5:E42;B\$48)

Donde:

- ☞ **D5:D42** es el rango que contiene los Montos destinados a cada obra,
- ☞ **B5:B42** es el rango donde buscará el primer criterio de selección,
- ☞ **A49** es el primer criterio de selección (también podía consignarse "Facultad de Bellas Artes" pero no se podría copiar hacia abajo),
- ☞ **E5:E42** es el rango donde buscará el segundo criterio de selección,
- ☞ **B\$48** es el segundo criterio de selección (el uso de referencia mixta es para poder copiar hacia abajo la función).

f3 =SUMAR.SI.CONJUNTO(D5:D42;B5:B42;A49;E5:E42;B\$48)					
	A	B	C	D	E
32	Adecuación Área Académica	Facultad de Cs. Jurídicas	Ministerio de Planificación de la Nación	\$ 262.527	En ejecución
33	Rampa de acceso para Discapacitados	Facultad de Humanidades	Presupuesto UNLP	\$ 20.700	En ejecución
34	Trabajos de Pintura	Bachillerato de Bellas Artes	Gobierno Pcia. Bs. As.	\$ 11.300	En proceso licitatorio
35	Liceo Victor Mercante Restauración Etapall	Liceo V. Mercante	Presidencia HCS	\$ 125.500	Para adjudicar
36	Ampliación Sector Administración	Facultad de Trabajo Social	Recursos de la Facultad	\$ 31.098	Terminado
37	Trabajos varios	Bachillerato de Bellas Artes	Gobierno Pcia. Bs. As.	\$ 13.501	Para adjudicar
38	Liceo Victor Mercante Restauración Etapall	Liceo V. Mercante	Presidencia HCS	\$ 749.500	En ejecución
39	Recambio de Cubiertas Aulas 6 y 7	Facultad de Trabajo Social	Recursos de la Facultad	\$ 58.200	Para adjudicar
40	Actualización de Circuitos eléctricos	Bachillerato de Bellas Artes	Gobierno Pcia. Bs. As.	\$ 7.980	En ejecución
41	Liceo Victor Mercante Restauración Etapall	Liceo V. Mercante	Presidencia HCS	\$ 174.057	En ejecución
42	Aula para conferencias	Facultad de Trabajo Social	Recursos de la Facultad	\$ 739.230	Terminado
43				\$ 6.960.735	
44					
45					
46					
47					
48	Unidad Académica	Terminado			
49	Facultad de Bellas Artes	\$ 174.877			
50	Facultad de Humanidades	\$ 4.230			
51	Facultad de Cs. Jurídicas	\$ 612.985			
52	Facultad de Cs. Astronómicas	\$ 1.896.002			
53		\$ 2.687.674			
54					
55					

Función CONTAR.SI.CONJUNTO

Permite contar frecuencia de sucesos que obedecen a **más de un criterio** simultáneamente.

Sintaxis =Contar.si.conjunto(rango_criterios1; criterio1; [rango_criterios2; criterio2]; ...)

A partir del rango_criterios2, es opcional. Se permiten hasta 127 pares de rango_criterios; criterio.

Ejemplo: De la misma planilla tratada para sumar.si.conjunto, extraeremos la cantidad de veces que la Universidad Nacional de La Plata fue la Fuente de Financiamiento para las obras de las Unidades Académicas especificadas.

f3 =CONTAR.SI.CONJUNTO(C5:C42;C\$48;B5:B42;A49)					
	A	B	C	D	E
28	Despachos Instituto de Física	Facultad de Cs. Exactas	CONICET	\$ 370.000	En ejecución
29	Construcción Grupo Sanitario	Facultad de Cs. Jurídicas	Ministerio de Educación de la Nación	\$ 14.950	En proceso licitatorio
30	Adecuación Instalación de gas	Facultad de Cs. Jurídicas	Presupuesto UNLP	\$ 6.500	En proceso licitatorio
31	Edificio Reforma Universitaria Etapa I	Facultad de Cs. Jurídicas	CONICET	\$ 162.350	Para adjudicar
32	Adecuación Área Académica	Facultad de Cs. Jurídicas	Ministerio de Planificación de la Nación	\$ 262.527	En ejecución
33	Rampa de acceso para Discapacitados	Facultad de Humanidades	Presupuesto UNLP	\$ 20.700	En ejecución
34	Trabajos de Pintura	Bachillerato de Bellas Artes	Gobierno Pcia. Bs. As.	\$ 11.300	En proceso licitatorio
35	Liceo Victor Mercante Restauración Etapall	Liceo V. Mercante	Presidencia HCS	\$ 125.500	Para adjudicar
36	Ampliación Sector Administración	Facultad de Trabajo Social	Recursos de la Facultad	\$ 31.098	Terminado
37	Trabajos varios	Bachillerato de Bellas Artes	Gobierno Pcia. Bs. As.	\$ 13.501	Para adjudicar
38	Liceo Victor Mercante Restauración Etapall	Liceo V. Mercante	Presidencia HCS	\$ 749.500	En ejecución
39	Recambio de Cubiertas Aulas 6 y 7	Facultad de Trabajo Social	Recursos de la Facultad	\$ 58.200	Para adjudicar
40	Actualización de Circuitos eléctricos	Bachillerato de Bellas Artes	Gobierno Pcia. Bs. As.	\$ 7.980	En ejecución
41	Liceo Victor Mercante Restauración Etapall	Liceo V. Mercante	Presidencia HCS	\$ 174.057	En ejecución
42	Aula para conferencias	Facultad de Trabajo Social	Recursos de la Facultad	\$ 739.230	Terminado
43				\$ 6.960.735	
44					
45					
46					
47					
48	Unidad Académica	Terminado	Presupuesto UNLP		
49	Facultad de Bellas Artes	\$ 174.877	3		
50	Facultad de Humanidades	\$ 4.230	4		
51	Facultad de Cs. Jurídicas	\$ 612.985	1		
52	Facultad de Cs. Astronómicas	\$ 1.896.002	0		
53		\$ 2.687.674	8		
54					

=Contar.si.conjunto(C5:C42;C\$48;B5:B42;A49)

Donde:

- ☞ **C5:C42** es el rango que contiene las Fuentes de Financiación de las Obras,
- ☞ **C\$48** es donde figura el ente de Financiación que nos interesa,
- ☞ **B5:B42** es el rango donde buscará el segundo criterio de selección,
- ☞ **A49** es el segundo criterio de selección (las Unidades Académicas de nuestro interés).

ANIDAMIENTO DE FUNCIONES

En algunas ocasiones, podemos necesitar utilizar una función como uno de los argumentos de otra función, es decir debemos anidar funciones. Una fórmula puede contener como máximo siete niveles de funciones anidadas.

A continuación presentamos, a modo de ejemplo, algunos casos:

Ejemplo 1

En la siguiente planilla figuran los promedios de los alumnos de Informática de tercer año, nos piden que en la columna C figure "Aprobado", si el promedio es mayor o igual a 6, "Diciembre", si el promedio es menor que 6 pero mayor o igual a 4, y "Marzo", si el promedio es menor que 4.

En este caso debemos utilizar como uno de los argumentos de la función SI otra función SI.

C7	=SI(B7>=6;"Aprobado";SI(B7>4;"Diciembre";"Marzo"))				
	A	B	C	D	E
1	INFORMÁTICA 3RO.C				
2					
3	ALUMNO	PROMEDIO	CONDICION		
4	Pérez, Juan	8	Aprobado		
5	Díaz, Lucía	5	Diciembre		
6	Ares, Esteban	3	Marzo		
7	Ríos, Griselda	2	Marzo		
8					
9					
10					

Ejemplo 2

En la siguiente planilla nos piden que en la columna C figure la leyenda "Inferior" si la cantidad de mm caídos, por provincia, es menor al promedio de lluvias de la zona especificada y "Superior o igual" en caso contrario.

C3 =SI(B3< PROMEDIO(\$B\$3:\$B\$9); "Inferior"; "Superior o igual")					
	A	B	C	D	E
1					
2	PROVINCIA	MM CAIDOS	En relación al promedio		
3	Corrientes	190	Superior o igual		
4	La Pampa	243	Superior o igual		
5	Chaco	38	Inferior		
6	Córdoba	75	Inferior		
7	Buenos Aires	64	Inferior		
8	Entre Ríos	104	Inferior		
9	Salta	58	Inferior		
10					

En este caso debemos utilizar como uno de los argumentos de la función SI la función Promedio.

Ejemplo 3

En la siguiente planilla nos piden que en la columna F figure la leyenda "Alerta de nevadas" si la temperatura registrada en cada localidad es inferior a la temperatura promedio y la humedad es inferior al 90 %, en caso contrario debemos dejar la celda en blanco.

F2 =SI(Y(D2<PROMEDIO(\$D\$2:\$D\$92);E2<90); "Alerta de nevadas"; " ")						
	A	B	C	D	E	F
1	Posición	Localidad	Provincia	Temperatura °C	Humedad	Pronóstico
2		BASE ORCADAS	ANTARTIDA	-0,6 °C	96%	Alerta de nevadas
3		BASE ESPERANZA	ANTARTIDA	-1,6 °C	78%	Alerta de nevadas
4		BASE MARAMBIO	ANTARTIDA	-10,0 °C	91%	Alerta de nevadas
5		BASE BELGRANO II	ANTARTIDA	-18,2 °C	51%	Alerta de nevadas
6		EZEIZA	BUENOS AIRES	13,4 °C	80%	
7		MORON	BUENOS AIRES	13,0 °C	82%	
8		EL PALOMAR	BUENOS AIRES	12,9 °C	83%	
9		SAN FERNANDO	BUENOS AIRES	12,2 °C	74%	
10		SAN MIGUEL	BUENOS AIRES	12,1 °C	85%	
11		PEHUAJO	BUENOS AIRES	11,6 °C	83%	
12		JUNIN	BUENOS AIRES	11,5 °C	88%	
13		BAHIA BLANCA	BUENOS AIRES	11,0 °C	89%	
14		NUEVE DE JULIO	BUENOS AIRES	10,9 °C	92%	
15		MAR DEL PLATA	BUENOS AIRES	10,2 °C	91%	
16		DOLORES	BUENOS AIRES	9,9 °C	91%	

En este caso debemos utilizar como uno de los argumentos de la función SI la función Promedio e Y.

Ejemplo 4

En la siguiente planilla nos piden que en la columna E figure el código A, B o C de acuerdo al siguiente criterio:

A: Para aquellos países que tengan una deuda externa menor a 40000 y cumplan con al menos una de las siguientes condiciones: PBI/Hab mayor que 4000 ó %Alfabet. mayor que 90.

B: Para los países con una deuda externa mayor o igual que 40000 y un PBI/Hab mayor que 4500

C: Para el resto de los casos.

E2		=SI(Y(B2<40000;O(C2>4000;D2>90));"A";SI(Y(B2>=40000;C2>4500);"B";"C"))				
	A	B	C	D	E	F
1	Pais	Deuda externa	PBI/Hab.	%Alfabet.	Codigo	
2	Argentina	155.000	5.000	97	B	
3	Brasil	182.000	3.850	85	C	
4	Chile	36.000	4.850	94	A	
5	Uruguay	21.000	4.700	97	A	
6	Bolivia	29.500	2.850	80	C	
7	Paraguay	32.000	2.850	80	C	
8	Perú	45.000	3.200	85	C	
9	Ecuador	56.000	3.000	83	C	
10	Venezuela	88.000	4.500	89	C	
11	Colombia	100.000	3.200	80	C	
12						

En este caso debemos utilizar como argumentos de la función SI las funciones: Y, O y SI.

Ejemplo 5

En la siguiente planilla nos piden que en la columna A figure el código de empleado. Dicho código se conforma de la siguiente manera:

El número de legajo, separado por un guion medio (-), le continúan las primeras tres letras correspondientes al cargo, nuevamente un guion bajo, y por último la letra M si el horario de entrada es a la 9 y la letra T en caso de que dicho horario sea a las 14.

A2		=CONCATENAR(B2;"_"&IZQUIERDA(D2;3);"_"&SI(E2=9;"M;"T"))				
	A	B	C	D	E	F
1	Código empleado	Legajo	Nombres de los empleados	Cargo	Horario Entrada	
2	1 Ger M	1	Agustín Perez	Gerente	9	
3	2 Sup M	2	Alejandro Ricci	Supervisor	9	
4	3 Coor T	3	Andrés Narvaez	Coordinador	14	
5	4 Sec T	4	Mariela Pirutti	Secretario	14	
6	5 Adm T	5	Luciano Mendez	Administrativo	14	
7	6 Sup M	6	Carolina Fal	Supervisor	9	
8	7 Coor T	7	Cristina Brites	Coordinador	14	
9	8 Adm T	8	Vanesa Lopez	Administrativo	14	
10	9 Adm T	9	Carlos Lopez	Administrativo	14	
11	10 Sec M	10	Mariano Garcia	Secretario	9	
12	11 Adm M	11	Pablo Rago	Administrativo	9	
13	12 Adm M	12	Florencia Peñaloza	Administrativo	9	
14	13 Sec T	13	Sebastián Bermudez	Secretario	14	
15	14 Adm T	14	Juan Calvo	Administrativo	14	

En este caso debemos utilizar como uno de los argumentos de la función Concatenar las funciones Izquierda y SI.

Funciones anidadas combinadas con cálculos.

En la siguiente planilla nos piden que a partir de la celda F19 figure el interés punitorio a pagar por los propietarios que abonan fuera de término las expensas correspondientes a su unidad funcional. Para ello debemos tener en cuenta la fecha de vencimiento (celda G16), la fecha de pago (E19), el interés punitorio (G14) que se cobra por cada día de mora sobre el importe del mes (D19).

F19	=SI(Y(MES(E19)>=MES(\$G\$16);DIA(E19)>10);SIFECHA(\$G\$16,E19;"d")*\$G\$14*D19;0)							
	A	B	C	D	E	F	G	H
14	TOTAL		4.163,10			Intereses Punitorio	0,50%	
15								
16	Facturación por unidad			Periodo:	Febrero	Vencimiento	10/03/2011	
17								
18	UF	Propietario	%	Importe del mes	Fecha Pago	Punitorio	Importe a pagar	
19	1	Maria Ovejero	2,20%	91,5882	04/03/2011	\$ 0,00	\$ 91,59	
20	2	Juan Pérez	8,00%	333,048	14/02/2011	\$ 0,00	\$ 333,05	
21	3	Sol López	7,80%	324,7218	18/02/2011	\$ 0,00	\$ 324,72	
22	4	Jorge Fernán	7,80%	324,7218	10/03/2011	\$ 0,00	\$ 324,72	
23	5	Mario Sogari	7,80%	324,7218	20/03/2011	\$ 16,24	\$ 340,96	
24	6	Nélida García	7,80%	324,7218	04/03/2011	\$ 0,00	\$ 324,72	
25	7	Lucía Ferraro	7,30%	303,9063	14/02/2011	\$ 0,00	\$ 303,91	
26	8	Carmen López	7,90%	328,8849	18/02/2011	\$ 0,00	\$ 328,88	
27	9	Hernán Crespo	8,20%	341,3742	10/03/2011	\$ 0,00	\$ 341,37	
28								

En este caso debemos utilizar como argumentos de la función SI las funciones: Y, MES, DIA y SIFECHA. Las operaciones que es necesario aplicar son: producto y porcentaje.

En la siguiente planilla nos piden calcular a partir de la celda **J4** el costo del flete, que surge de aplicar un cierto porcentaje sobre el Total bruto (Columna I), teniendo en cuenta el rubro al cual pertenece el artículo, las unidades vendidas y el total bruto, para lo cual debemos atender las siguientes consideraciones:

Si la cantidad de unidades vendidas supera las 500 unidades y corresponde al rubro Cerámica, el costo de flete es un 15% del Total bruto.

Si en cambio, la cantidad de unidades vendidas se encuentra superior a la cantidad de 100, pero es inferior a las 500, el costo resultará de aplicar un 10% sobre el total bruto. En caso contrario se aplicará sólo un 5% sobre el mismo concepto.

J4	=SI(Y(H4>500;E4="CERAMICA");15%*I4;SI(Y(H4>100;H4<500);10%*I4;5%*I4))							
	C	E	F	G	H	I	J	K
1								
2								
3	Título	Rubro	Precio Unitario	Descuento x Rubro	Unidades Vendidas	Total Bruto	Costo Flete	Total Facturado
4	CERAMICA, PASTAS Y VIDRIADOS	CERAMICA	\$ 30,00	\$ 3,00	600	16200	2430	18630
5	TECNICAS DE DECORACION CERAMICA	CERAMICA	\$ 76,00	\$ 7,60	600	41040	6156	47196
6	FORMULARIO Y PRACTICAS DE CERAMICA	CERAMICA	\$ 43,00	\$ 4,30	350	13545	1354,5	14899,5
7	MANUAL PRACTICO DE CERAMICA	CERAMICA	\$ 76,00	\$ 7,60	45	3078	153,9	3231,9
8	CERAMICA VIVA	CERAMICA	\$ 76,00	\$ 7,60	10	684	34,2	718,2
9	RECETAS BARNICES PARA CERAMISTAS	CERAMICA	\$ 40,00	\$ 4,00	450	16200	1620	17820
10	VIDRIADOS CERAMICOS	CERAMICA	\$ 99,00	\$ 9,90	70	6237	311,85	6548,85
11	LA SERIGRAFIA EN LA CERAMICA	CERAMICA	\$ 74,00	\$ 7,40	120	7992	799,2	8791,2
12	INTROD TECNOLOGIA DE LA CERAMICA	CERAMICA	\$ 61,00	\$ 6,10	45	2470,5	123,525	2594,025
13	PRODUCTOS CERAMICOS COCCION	CERAMICA	\$ 70,00	\$ 7,00	550	34650	5197,5	39847,5
14	TECNICA DE LA CERAMICA	CERAMICA	\$ 58,00	\$ 5,80	540	28188	4228,2	32416,2
15	TRATADO DE CERAMICA	CERAMICA	\$ 63,00	\$ 6,30	230	13041	1304,1	14345,1

En este caso debemos utilizar como argumentos de la función SI las funciones: Y y SI.

La operación que es preciso aplicar es: porcentaje.